

11. Conditional and Repeated Statements and Boolean Expressions

11.1. Comparisons and jumps

Conditional and repeated statements are implemented with the aid of jump instructions, also called *branch* instructions. As a first example, let us consider the simplest form of conditional statement:

```
IF x = y THEN StatSequence END
```

Its mapping into a sequence of instructions is straightforward:

```
IF x = y          EQL x, y
                  BF  L
                  code(StatSequence)
THEN StatSequence
END              L  ...
```

Our considerations are based once again on a stack architecture. Instruction EQL tests the two operands for equality and replaces them on the stack by the Boolean result. The subsequent branch instruction BF (branch if FALSE) leads to the destination label L if this result is FALSE, and removes it from the stack. Similarly to EQL, conditional branch instructions are postulated for the relations $=$, $<$, $>$, \leq , \geq , and \neq .

Unfortunately, however, such compiler-friendly computers are hardly widespread. Rather more common are computers whose branch instructions depend on the comparison of a register value with 0. We denote them as BNE (branch if not equal), BLT (branch if less than), BGE (branch if greater or equal), BLE (branch if less or equal), and BGT (branch if greater than). The code sequence corresponding to the above example is

```
IF x = y          code (Ri := x - y)
                  BNE Ri, L
                  code(StatSequence)
THEN StatSequence
END              L  ...
```

The use of subtraction ($x - y \geq 0$ standing for $x \geq y$) has an implicit pitfall: subtraction may lead to overflow, resulting in either program termination or a wrong result. Therefore a specific comparison instruction CMP is used in place of subtraction, which avoids overflow, but correctly indicates whether the difference is either zero, positive or negative. The result is typically stored in a special register called condition code, consisting of the two bits denoted by N and Z, indicating whether the difference is negative or zero respectively. All conditional branch instructions then implicitly refer to this register as argument.

```
IF x = y          CMP x, y
                  BNE L
                  code(StatSequence)
THEN StatSequence
END              L  ...
```

11.2. Conditional and repeated statements

The question of how a Boolean value is to be represented by an item now arises. In the case of stack architecture the answer is easy: since the result of a comparison lies on the stack like any other result, no special item mode is necessary. A CMP instruction, however, requires further thought. We shall first restrict our consideration to the simple cases of pure comparisons without further Boolean operators.

In the case of an architecture with a CMP scheme, it is necessary to indicate in the resulting item which register holds the computed difference, and which relation is represented by the comparison. For the latter a new attribute is required; we call the new mode *Cond* and its new attribute (record field) *c*. The mapping of relations to values of *c* is defined by

=	0	#	1
<	2	>=	3
<=	4	>	5

The construct containing comparisons is the expression. Its syntax is

expression = SimpleExpression [("=" | "#" | "<" | "<=" | ">" | ">=") SimpleExpression].

The corresponding, extended parser procedure is easily derived:

```

PROCEDURE expression(VAR x: Item);
  VAR y: Item; op: INTEGER;
BEGIN SimpleExpression(x);
  IF (sym >= eql) & (sym <= gtr) THEN
    op := sym; Get(sym); SimpleExpression(y); Relation(op, x, y)
  END
END expression;

PROCEDURE Relation(op: INTEGER; VAR x, y: Item);
BEGIN
  IF (x.type.form # Integer) OR (y.type.form # Integer) THEN Mark("bad type")
  ELSE
    IF (y.mode = Const) & (y.a = 0) THEN load(x) ELSE PutOp(CMP, x, y) END ;
    x.c := op - eql; EXCL(regs, x.r); EXCL(regs, y.r)
  END ;
  x.mode := Cond; x.type := boolType
END Relation;

```

The code scheme presented at the beginning of this chapter yields the corresponding parser program for handling the IF construct in *StatSequence*:

```

ELSIF sym = if THEN
  Get(sym); expression(x); CJump(x);
  IF sym = then THEN Get(sym) ELSE Mark("THEN?") END ;
  StatSequence; Fixup(x.a)
  IF sym = end THEN Get(sym) ELSE Mark("END?") END

```

Procedure *CJump(x)* generates the necessary branch instruction according to its parameter *x.c* in such a way that the jump is taken if the specified condition is not satisfied.

Here a difficulty becomes apparent which is inherent in all single-pass compilers. The destination location of branches is still unknown when the instruction is to be emitted. This problem is solved by adding the location of the branch instruction as an attribute to the item generated. This attribute is used later when the destination of the jump becomes known in order to complete the branch with its true address. This is called a *fixup*. The simple solution is possible only if code is deposited in a global array where elements are accessible at any time. It is not applicable if the emitted code is immediately stored on disk. To represent the address of the incomplete branch instruction we use the item field *a*.

```

PROCEDURE CJump(VAR x: Item);
BEGIN
  IF x.type.form = Boolean THEN
    Put(BEQ + negated(x.c), x.r, 0, 0); EXCL(regs, x.r); x.a := pc-1
  ELSE OSS.Mark("Boolean?"); x.a := pc
  END
END CJump;

PROCEDURE negated(cond: LONGINT): LONGINT;
BEGIN
  IF ODD(cond) THEN RETURN cond-1 ELSE RETURN cond+1 END
END negated;

PROCEDURE Fixup(L: LONGINT);
BEGIN code[L] := code[L] DIV 10000H * 10000H + pc - L
END Fixup;

```

Procedure *CJump* issues an error message if x is not of type BOOLEAN. Note that branch instructions use addresses relative to the instruction's location (PC-relative); therefore the value $pc-L$ is used.

Finally, we have to show how conditional statements in their general form are compiled; the syntax is

```
"IF" expression "THEN" StatSequence
{"ELSIF" expression "THEN" StatSequence}
["ELSE" StatSequence]
"END"
```

and the corresponding code pattern is

IF expression THEN		code(expression)
		Bcond L0
StatSequence		code(StatSequence)
		BR L
ELSIF expression THEN	L0	code(expression)
		Bcond L1
StatSequence		code(StatSequence)
		BR L
ELSIF expression THEN	L1	code(expression)
		Bcond L2
StatSequence		code(StatSequence)
		BR L
.....		
ELSE StatSequence	Ln	code(StatSequence)
END	L	...

from which the parser statements can be derived as part of procedure *StatSequence*. Although an arbitrary number of ELSIF constructs can occur and thereby also an arbitrary number of jump destinations $L1$, $L2$, ... may result, a single item variable x suffices. It is assigned a new value for every ELSIF instance.

```
ELSIF sym = if THEN
  Get(sym); expression(x); CJump(x);
  IF sym = then THEN Get(sym) ELSE Mark("THEN ?") END ;
  StatSequence; L := 0;
  WHILE sym = elsif DO
    Get(sym); FJump(L); Fixup(x.a); expression(x); CJump(x);
    IF sym = then THEN Get(sym) ELSE Mark("THEN ?") END ;
    StatSequence
  END ;
  IF sym = else THEN Get(sym); FJump(L); Fixup(x.a); StatSequence ELSE Fixup(x.a) END ;
  FixLink(L);
  IF sym = end THEN Get(sym) ELSE Mark("END ?") END
  ...

PROCEDURE FJump(VAR L: LONGINT);
BEGIN Put(BEQ, 0, 0, L); L := pc-1
END FJump
```

However, a new situation arises in which not only a single branch refers to the destination label L at the end, but an entire set, namely as many as there are IF and ELSIF branches in the statement. The problem is elegantly solved by storing the links of the list of incomplete branch instructions in these instructions themselves, and to let variable L represent the root of this list. The links are established by the parameter of the *Put* operation called in *FJump*. It suffices to replace procedure *Fixup* by *FixLink*, in which the entire list of instructions to be fixed up is traversed. It is essential that variable L is declared local to the parser procedure *StatSequence*, because statements may be nested, which leads to recursive activation. In this case, several instances of variable L coexist representing different lists.

```

PROCEDURE FixLink(L: LONGINT);
  VAR L1: LONGINT;
BEGIN
  WHILE L # 0 DO
    L1 := code[L] MOD 10000H; Fixup(L); L := L1
  END
END FixLink;

```

Compilation of the WHILE statement is very similar to that of the simple IF statement. In addition to the conditional forward jump, an unconditional backward jump is necessary. The syntax and the corresponding code pattern are:

WHILE expression DO	L0 code(expression)
	Bcond L1
StatSequence	code(StatSequence)
END	BR L0
	L1 ...

From this we derive the corresponding, extended parser procedure:

```

ELSIF sym = while THEN
  Get(sym); L := pc; expression(x); CJump(x);
  IF sym = do THEN Get(sym) ELSE Mark("DO ?") END ;
  StatSequence; BJump(L); Fixup(x.a);
  IF sym = end THEN Get(sym) ELSE Mark("END ?") END
PROCEDURE BJump(L: LONGINT);
BEGIN Put(BEQ, 0, 0, L-pc)
END BJump;

```

To summarize, we display two statements using variables *i* and *j*, together with the generated code:

```

IF i < j THEN i := 0 ELSIF i = j THEN i := 1 ELSE i := 2 END ;
WHILE i > 0 DO i := i - 1 END

```

4	LDW	0, base, -4	i
8	LDW	1, base, -8	j
12	CMP	0, 0, 1	
16	BGE	3	(jump over 3 instructions to 28)
20	STW	0, base, -4	i := 0
24	BEQ	10	(jump over 10 instructions to 64)
28	LDW	0, base -4	
32	LDW	1, base, -8	
36	CMP	0, 0, 1	
40	BNE	4	(jump over 4 instructions to 56)
44	MOVI	0, 0, 1	
48	STW	0, base, -4	i := 1
52	BEQ	3	(jump over 3 instructions to 64)
56	MOVI	0, 0, 2	
60	STW	0, base, -4	i := 2
64	LDW	0, base, -4	
68	BLE	5	(jump over 5 instructions to 88)
72	LDW	0, base, -4	
76	SUBI	0, 0, 1	
80	STW	0, base, -4	i := i - 1
84	BEQ	-5	(jump back over 5 instructions to 64)
88	...		

11.3. Boolean operations

It is of course tempting to treat Boolean expressions in the same way as arithmetic expressions. Unfortunately, however, this would in many cases lead not only to inefficient, but even to wrong code. The reason lies in the definition of Boolean operators, namely

$p \text{ OR } q = \text{if } p \text{ then TRUE else } q$
 $p \text{ \& } q = \text{if } p \text{ then } q \text{ else FALSE}$

This definition specifies that the second operand q need not be evaluated if the result is uniquely given by the value of the first operand p . Programming language definitions even go a step further by specifying that in these cases the second operand must not be evaluated. This rule is postulated in order that the second operand may be left undefined without causing program execution to be terminated. A frequent example involving a pointer x is

$(x \# \text{NIL}) \text{ \& } (x.^{\wedge}.\text{size} > 4)$

Boolean expressions with Boolean operators therefore assume the form of conditional statements (more precisely, conditional expressions), and it is appropriate to use the same compilation techniques as for conditional statements. Boolean expressions and conditional statements merge, as the following example shows. The statement

IF $(x \leq y) \text{ \& } (y < z)$ THEN S END

is compiled in the same way as its equivalent formulation

IF $x \leq y$ THEN IF $y < z$ THEN S END END

With the intention of deriving a suitable code pattern, let us first consider the following expression containing three relations connected by the $\&$ operator. We postulate the desired code pattern as shown below, considering only the pattern to the left for the moment. a, b, \dots, f denote numeric values. The labels T and F denote the destinations for the cases when the expression is true or false, respectively.

$(a < b) \text{ \& } (c < d) \text{ \& } (e < f)$	
CMP a, b	CMP a, b
BGE F	BGE F
CMP c, d	CMP c, d
BGE F	BGE F
CMP e, f	CMP e, f
BGE F	BLT T
(T)	(F)

As the left hand pattern shows, a conditional branch instruction is emitted for every $\&$ operator. The jump is executed if the preceding condition is not satisfied (F-jump). This results in the instructions BGE to represent the $<$ relation, BNE for the $=$ relation, and so on.

If we consider the problem of generating the required code, we can see that the parser procedure term, as it is known for processing arithmetic terms, must be extended slightly. In particular, a branch instruction must be emitted before the second operand is processed, whereas at the end this instruction's address must be fixed up. The former task is performed by procedure *Op1*, the latter by *Op2*.

```
PROCEDURE term(VAR x: Item);
  VAR y: Item; op: INTEGER;
  BEGIN factor(x);
    WHILE (sym >= times) & (sym <= and) DO
      op := sym; Get(sym);
      IF op = and THEN Op1(op, x) END ;
      factor(y); Op2(op, x, y)
    END
  END term;
```

```

PROCEDURE Op1(op: INTEGER; VAR x: Item); (* x := op x *)
  VAR t: LONGINT;
BEGIN
  IF op = minus THEN ...
  ELSIF op = and THEN
    IF x.mode # Cond THEN loadBool(x) END ;
    PutBR(BEQ + negated(x.c), x.a); EXCL(regs, x.r); x.a := pc-1
  END
END Op1;

```

If the first Boolean factor is represented by item x in mode *Cond*, then at the present position x is TRUE and the instructions for the evaluation of the second operand must follow. However, if it is not in mode *Cond*, it must be transferred into this mode. This task is executed by procedure *loadBool*. We assume that the value FALSE is represented by 0. The attribute value $c = 1$ therefore causes the instruction BEQ to become active, if x equals 0.

```

PROCEDURE loadBool(VAR x: Item);
BEGIN
  IF x.type.form # Boolean THEN OSS.Mark("Boolean?") END ;
  load(x); x.mode := Cond; x.c := 1
END loadBool;

```

The OR operator is treated analogously, with the difference that jumps are taken if their respective conditions are satisfied (T-jump). The instructions are listed in the dual list with links in the item field b . The postcondition of a sequence of terms connected with OR operators is FALSE. Consider again the left-hand code pattern only:

(a < b) OR (c < d) OR (e < f)	
CMP a, b	CMP a, b
BLT T	BLT T
CMP c, d	CMP c, d
BLT T	BLT T
CMP e, f	CMP e, f
BLT T	BGE F
(F)	(T)

Next, we consider the implementation of negation. Here it turns out that under the scheme presented no instructions need be emitted whatsoever. Only the condition value represented by the item field c has to be negated, and the lists of F-jumps and T-jumps need be exchanged. The result of negation is shown in the code patterns in Figures 11.1 and 11.2 on the right-hand side for both expressions with & and OR operators. The affected procedures are extended as shown below:

```

PROCEDURE SimpleExpression(VAR x: Item);
  VAR y: Item; op: INTEGER;
BEGIN term(x);
  WHILE (sym >= plus) & (sym <= or) DO
    op := sym; Get(sym);
    IF op = or THEN Op1(op, x) END ;
    term(y); Op2(op, x, y)
  END
END SimpleExpression;

PROCEDURE Op1(op: INTEGER; VAR x: Item); (* x := op x *)
  VAR t: LONGINT;
BEGIN
  IF op = minus THEN ...
  ELSIF op = not THEN
    IF x.mode # Cond THEN loadBool(x) END ;
    x.c := negated(x.c); t := x.a; x.a := x.b; x.b := t

```

```

ELSIF op = and THEN
  IF x.mode # Cond THEN loadBool(x) END ;
  PutBR(BEQ + negated(x.c), x.a); EXCL(regs, x.r);
  x.a := pc-1; FixLink(x.b); x.b := 0
ELSIF op = or THEN
  IF x.mode # Cond THEN loadBool(x) END ;
  PutBR(BEQ + x.c, x.b); EXCL(regs, x.r);
  x.b := pc-1; FixLink(x.a); x.a := 0
END
END Op1;

```

When compiling expressions with & and OR operators, care must be taken that in front of every & condition P , and in front of every OR condition $\sim P$, must hold. The respective lists of jump instructions must be traversed (the T-list for &, the F-list for OR), and the designated instructions must be fixed up appropriately. This occurs through procedure calls of *FixLink* in *Op1*. As examples, we consider the expressions

```

(a < b) & (c < d) OR ((e < f) & (g < h))
(a < b) OR (c < d) & ((e < f) OR (g < h))

```

and the resulting codes:

	CMP a, b		CMP a, b
	BGE F0		BLT T0
	CMP c, d		CMP c, d
	BLT T		BGE F
F0	CMP e, f	T0	CMP e, f
	BGE F		BLT T
	CMP g, h		CMP g, h
	BGE F		BGE F
	(T)		(T)

It may also happen that a list of a subordinate expression may merge with the list of its containing expression (see F-link in the pattern for Q in Figure 11.3). This merger is accomplished by procedure *merged(a, b)*, yielding as its value the concatenation of its argument lists. It is called from within procedure *Op2*.

```

PROCEDURE Op2(op: INTEGER; VAR x, y: Item); (* x := x op y *)
BEGIN
  IF (x.type.form = Integer) & (y.type.form = Integer) THEN
    ...
  ELSIF (x.type.form = Boolean) & (y.type.form = Boolean) THEN
    IF y.mode # Cond THEN loadBool(y) END ;
    IF op = or THEN x.a := y.a; x.b := merged(y.b, x.b); x.c := y.c
    ELSIF op = and THEN x.a := merged(y.a, x.a); x.b := y.b; x.c := y.c
    END
  ELSE ...
  END ;
END Op2;

```

11.4. Assignments to Boolean variables

Compilation of an assignment to a Boolean variable q is certainly more complicated than commonly expected. The reason is the item mode *Cond*, which must be converted into an assignable value 0 or 1. This is achieved by the following code pattern:

```

T   ADDI  0, 0, 1
    BEQ   L
F   ADDI  0, 0, 0

```

```
L      STW      0, q
```

This causes the simple assignment $q := x < y$ to appear as a disappointingly long code sequence. We should, however, be aware that Boolean variables (commonly called *flags*) occur (should occur) infrequently, although the notion of the type *Boolean* is indeed fundamental. It is inappropriate to strive for optimal implementation of rarely occurring constructs at the price of an intricate process. However, it is essential that the frequent cases are handled optimally.

Nevertheless, we handle assignments to a Boolean item not in the *Cond* mode as a special case, namely as a conventional assignment avoiding the involvement of jumps. Hence, the assignment $p := q$ results in the expected code sequence

```
LDW      1, 0, q
STW      1, 0, p
```

As a consequence, the affected procedure *Store* turns out as follows:

```
PROCEDURE Store(VAR x, y: Item); (* x := y *)
BEGIN ...
  IF y.mode = Cond THEN
    FixLink(y.b); GetReg(y.r); Put(MOVI, y.r, 0, 1); PutBR(BEQ, 2);
    FixLink(y.a); Put(MOVI, y.r, 0, 0)
  ELSIF y.mode # Reg THEN load(y)
  END ;
  IF x.mode = Var THEN Put(STW, y.r, x.r, x.a)
  ELSE Mark("illegal assignment")
  END ;
  EXCL(regs, x.r); EXCL(regs, y.r)
END Store;
```

11.5. Exercises

11.1. Mutate the language Oberon-0 into a variant Oberon-D by redefining the conditional and the repeated statement as follows:

```
statement = ...
  "IF" guardedStatements {"|" guardedStatements} "FI" |
  "DO" guardedStatements {"|" guardedStatements} "OD" .
guardedStatements = condition "." statement {";" statement} .
```

The new form of statement

```
IF B0 . S0 | B1 . S1 | ... | Bn . Sn FI
```

shall mean that of all conditions (Boolean expressions) B_i that are true, one is selected arbitrarily and its corresponding statement sequence S_i is executed. If none is true, program execution is aborted. Any statement sequence S_i will be executed only when the corresponding condition B_i is true. B_i is therefore said to be the guard of S_i .

The statement

```
DO B0 . S0 | B1 . S1 | ... | Bn . Sn OD
```

shall mean that that as long as any of the conditions B_i is true, one of them is chosen arbitrarily, and its corresponding statement sequence S_i is executed. The process terminates as soon as all B_i are false. Here too, the B_i function as guards. The DO-OD construct is a repetitive, nondeterministic construct. Adjust the compiler accordingly.

11.2. Extend Oberon-0 and its compiler by a FOR statement:

```
statement = [assignment | ProcedureCall |
  IfStatement | WhileStatement | ForStatement.
```


ForStatement = "FOR" identifier ":" expression "TO" expression ["BY" expression]
"DO" StatementSequence "END" .

The expression preceding the symbol TO specifies the starting value, the one thereafter the ending value of the control variable denoted by the identifier. The expression after BY indicates the increment. If missing, let 1 be its default value.

11.3. Consider the implementation of the case statement of Oberon (see Appendix A.2). Its essential property is that it uses a table of jump addresses for the various cases, and an indexed jump instruction.

12. Procedures and the Concept of Locality

12.1. Run_time organization of the store

Procedures, which are also known as subroutines, are perhaps the most important tool for structuring programs. Because of their frequency of occurrence, it is mandatory that their implementation is efficient. Implementation is based on the branch instruction which saves the current PC value and thereby the point of return after termination of the procedure, when this value is reloaded into the PC register.

The question as to where the return address should be saved arises immediately. In many computers it is deposited in a register, and we have adopted this solution in our RISC. This guarantees the utmost efficiency, because no additional memory access is involved. But having to save the register's value into memory before the next procedure call is unavoidable, because otherwise the old return address would be overwritten. Thereby the return address of the first call would be lost. In the implementation of a compiler this *link register* value must be saved at the beginning of each procedure call.

To store the link, a stack is the obvious solution. The reason is that procedure activations occur in a nested fashion; procedures terminate in the reverse order of their calls. The store for the return addresses must therefore operate according to the *first-in last-out* principle. This results in the following, fixed code sequences at the beginning and end of every procedure. They are called the procedure's *prologue* and *epilogue*. Here we will use R13 for the stack pointer SP and R14 as link register LNK. R15 is defined as the program counter PC.

Call	BSR P	branch to subroutine
Prologue	P PSH LNK, SP, 4	push link
Epilogue	POP LNK, SP, 4	pop link
	RET LNK	return jump

This code pattern is valid under the assumption that the BSR instruction deposits the return address in R14. Note that this is specified as a hardware feature (Chapter 9), whereas the use of R13 as stack pointer is merely a software convention determined by the compiler design or by the underlying operating system. Whenever the system is started, R14 must be initialized to point to an area of memory reserved for the stack.

Algol 60 introduced the very fundamental concept of *local* variables. It implied that every identifier declared had a limited range of visibility and validity. In Pascal (and also in Oberon) this range is the procedure body. In concrete terms, variables may be declared local to a procedure such that they are visible and valid within this procedure only. The intended consequence is that upon entry to the procedure memory is allocated automatically for these local variables, and it is released upon the procedure's termination. Local variables of different procedures may therefore share the same storage area, but never simultaneously, of course.

At first sight this scheme seems to inflict a certain loss of efficiency upon the procedure call mechanism. Fortunately, however, this need not be so, because the storage blocks for the sets of local variables can be allocated, like return addresses, according to the stack principle. The return address may indeed also be considered as a (hidden) local variable, and it is only natural to use the same stack for variables and return addresses. The storage blocks are called *procedure activation records* or *activation frames*. Release of a block upon procedure termination is achieved by simply resetting the stack pointer to its value before the procedure call. Hence, allocation and release of local storage is optimally efficient.

Addresses of local variables generated by the compiler are always relative to the base address of the respective activation frame. Since in programs most variables are local, their addressing also must be highly efficient. This is achieved by reserving a register to hold the base address, and to make use of the fact that the effective address is the sum of a register value and the instruction's address field (register relative addressing mode). The reserved register is called the *frame pointer* (FP). These considerations

are taken into account by the following prologue and epilogue, where R12 assumes the role of the frame pointer:

Prolog	P	PSH	LNK, SP, 4	push link
		PSH	FP, SP, 4	push FP
		MOV	FP, 0, SP	FP := SP
		SUBI	SP, SP, n	SP := SP-n (n = frame size)
Epilog		MOV	SP, 0, FP	SP := FP
		POP	FP, SP, 4	pop FP
		POP	LNK, SP, 4	pop link
		RET	LNK	return jump

The activation frames resulting from consecutive procedure calls are linked by a list of their base addresses. The list is called the *dynamic link*, because it denotes the dynamic sequence of procedure activations. Its root lies in the frame pointer register FP (see Figure 12.1).




Figure 12.1. List of activation frames in the stack.

The state of the stack before and after a procedure call is shown in Figure 12.2. Note that the epilogue reverts the stack to its original state by removing return address and dynamic link entry.




Figure 12.2. States of the stack before and after procedure call.

If we carefully consider the necessity of the two pointers SP and FP, we may come to the conclusion that FP is actually superfluous, because the variables' offset addresses could be made relative to SP instead of FP. This proposition, however, is valid only if the sizes of all variables are known at compile time. This is not so in the case of open (dynamic) arrays, as will become apparent later. But obviously the retention

of a second pointer (FP) requires additional storage accesses upon every procedure call and return, which are undesirable.

In order to improve efficiency and in particular to reduce the length of the instruction sequences for both prologue and epilogue, computers with more complex instructions feature special instructions corresponding to prologue and epilogue. Two examples may help at this point; the second features special, dedicated registers for the pointers SP and FP. The number of required instructions, however, remains the same.

	Motorola 680x0	National Semiconductor 32x32
Call	BSR P	BSR P
Prologue	LINK D14, n	ENTER n
Epilogue	UNLNK D14	EXIT
Return jump	RTD	RET

12.2. Addressing of variables

We recall that the address of a local variable is relative to the base address of the activation frame containing the variable, and that this base address is held in register FP. The latter, however, holds only for the record activated last, and thereby only for variables which belong to the procedure in which they are referenced. In many programming languages procedure declarations may be nested, giving rise to references to variables which are local to some procedure, but not to the procedure referencing them. The following example demonstrates the situation, with R being local to Q, and Q and S local to P:

	Object	Level
PROCEDURE P;	P	0
VAR x: INTEGER;	x	1
PROCEDURE Q;	Q	1
VAR y: INTEGER;	y	2
PROCEDURE R;	R	2
VAR z: INTEGER;	z	3
BEGIN x := y + z		
END R;		
BEGIN R		
END Q ;		
PROCEDURE S;	S	1
BEGIN Q		
END S;		
BEGIN Q; S		
END P;		

Let us trace the chain of calls $P \rightarrow Q \rightarrow R$. It is tempting to believe that, when accessing variables x , y , or z in R , their base address could be obtained by traversing the dynamic link list. The number of steps would be the difference between the levels of the call and of the declaration. This difference is 2 for x , 1 for y , and 0 for z . But this assumption is wrong. R could also be reached through the call sequence $P \rightarrow S \rightarrow Q \rightarrow R$ as shown in Figure 12.3. Access to x would then lead in two steps to the activation frame of S instead of P .

Evidently, a second list of activation records is necessary which mirrors the static order of nesting rather than the dynamic order of calls. Hence a second link must be established upon every procedure call. The so-called *procedure mark* now contains, in addition to the return address and the dynamic link, a *static link* element. The static link of a procedure P points to the activation record of the procedure which contains P , that is, in which P is declared locally. It should be noted that this pointer is superfluous for procedures declared globally, if global variables are addressed directly, that is, without base address. Since this is typically the case, and since most procedures are declared globally, the additional

complexity caused by the static chain is acceptable. With some justification the absolute addressing of global variables can be considered as a special case of local variable addressing leading to an increase in efficiency.




Figure 12.3. Dynamic and static links in the stack.

Finally, note that access to variables via the static link list (intermediate level variables) is less efficient than access to strictly local variables, because every step through the list requires an additional memory access. Several solutions have been proposed and implemented to eliminate this loss of efficiency. They ultimately always rely on the mapping of the static list onto a set of base registers. We consider this as an optimization at the wrong place. First, registers are scarce resources which should not be given away too easily. And second, the copying of link elements into registers upon every call and return may easily cost more than it saves, in particular because references to intermediate-level variables occur quite rarely in practice. The optimization may therefore turn out to be quite the reverse.

In order not to enhance the readability of the Oberon-0 compiler listed in Appendix C, the handling of intermediate-level variables has not been implemented.

Global variables have fixed addresses which must also be considered relative to a frame address. Their absolute values are determined upon loading the code, that is, *after* compilation but *before* program execution. The emitted object code can therefore be accompanied by a list of addresses of instructions referring to global variables. The loader must then add to these addresses the base address of the respective frame of global variables. This fixup operation can be omitted if the computer features the program counter as an address register. Our RISC does exactly this by letting the PC be accessible as R15. The frame of global variables is placed immediately preceding the code frame. Hence, addresses of global variables use R15 as base address, and the address of the current instruction must be subtracted from the variable's offset.

12.3. Parameters

Parameters constitute the interface between the calling and the called procedures. Parameters on the calling side are said to be *actual parameters*, and those on the called side *formal parameters*. The latter are in fact only place holders for which the actual parameters are substituted. Basically, a substitution is an assignment of the actual value to the formal variable. This implies that every formal parameter be represented by a variable bound to the procedure, and that every call be accompanied by a number of assignments called parameter substitutions.

Most programming languages distinguish between at least two kinds of parameters. The first is the *value parameter* where, as its name suggests, the value of the actual parameter is assigned to the formal variable. The actual parameter is syntactically an expression. The second kind of parameter is the *reference parameter*, where, also as suggested by its name, a reference to the actual parameter is assigned to the formal variable. Evidently, the actual parameter must in this case be a variable, because an assignment to the formal parameter is permissible, and this assignment must refer to the actual variable.

(In Pascal, Modula, and Oberon, the reference parameter is therefore called variable parameter). The value of the formal variable is in this case a hidden pointer, that is, an address.

Of course, the actual parameter must be evaluated before the substitution takes place. In the case of variable parameters, the evaluation takes the form of an identification of the variable, implying, for example, the evaluation of the index in the case of indexed variables. But how is the destination of this substitution to be determined? Here the stack organization of the store comes into play. The actual values are simply deposited in sequence on the top of the stack; no explicit destination addresses are required. Figure 12.4 shows the state of the stack after the deposition of the parameters, and after the call and the prologue.




Figure 12.4. Parameter substitution.

It now becomes evident that parameters can be addressed relative to the frame address FP, like local variables. If local variables have negative offsets, parameters have positive offsets. It is particularly worth noting that the called procedure references parameters exactly where they were deposited by the calling procedure. The space allocated for the parameters is regained by the epilogue simply by increasing the value of SP.

Epilogue	MOV SP, 0, FP	SP := FP
	POP FP, SP, 4	pop FP
	POP LNK, SP, m+4	pop link and parameters
	RET LNK	return jump

In the case of CISC computers with prologue and epilogue represented by special instructions, the required increment of SP is included in the return instruction specifying the size of the parameter block as parameter (RET m).

12.4. Procedure declarations and calls

The procedure for processing procedure declarations is easily derived from the syntax with the aid of the parser construction rules. The new entry in the symbol table generated for a procedure declaration obtains the class attribute *Proc*, and its attribute *a* is given the current value of *pc*, which is the entry address of the procedure's prologue. Thereafter, a new scope is opened in the symbol table in such a way that (1) new entries for local objects are automatically inserted in the new scope, and (2) at the end of the procedure the local objects are easily discarded and the previous scope reappears. Here too, the two procedures *OpenScope* and *CloseScope* embody the stack principle, and the linkage is established by a header element (class *Head*, field *dsc*). Objects are given an additional attribute *lev* denoting the nesting level of the declared object. Consider the following declarations:

```
CONST N = 10;
VAR x: T;
PROCEDURE P(x, y: INTEGER); ...
```

The resulting symbol table is shown in Figure 12.5. The *dsc* pointer refers to *P*'s parameters *x* and *y*.




Figure 12.5. Symbol table representing two scopes.

```

PROCEDURE ProcedureDecl;
VAR proc, obj: Object;
    procid: Ident;
    locblksize, parblksize: LONGINT;

PROCEDURE FPSection;
    VAR obj, first: Object; tp: Type; parsize: LONGINT;
BEGIN
    IF sym = var THEN Get(sym); IdentList(Par, first) ELSE IdentList(Var, first) END ;
    IF sym = ident THEN
        find(obj); Get(sym);
        IF obj.class = Typ THEN tp := obj.type ELSE Mark("type?"); tp := intType END
    ELSE Mark("ident?"); tp := intType
    END ;
    IF first.class = Var THEN parsize := tp.size ELSE parsize := 4 END ;
    obj := first;
    WHILE obj # guard DO obj.type := tp; INC(parblksize, parsize); obj := obj.next END
END FPSection;

BEGIN (* ProcedureDecl *) Get(sym);
    IF sym = ident THEN
        procid := id;
        NewObj(proc, Proc); Get(sym); parblksize := 8;
        INC(level); OpenScope; proc.val := -1;
        IF sym = lparen THEN Get(sym);
            IF sym = rparen THEN Get(sym)
            ELSE FPSection;
                WHILE sym = semicolon DO Get(sym); FPSection END ;
                IF sym = rparen THEN Get(sym) ELSE Mark(")?") END
            END
        END ;
        obj := topScope.next; locblksize := parblksize;
        WHILE obj # guard DO

```

```

    obj.lev := curlev;
    IF obj.class = Par THEN
        DEC(locblksize, 4) ELSE locblksize := locblksize - obj.type.size
    END ;
    obj.val := locblksize; obj := obj.next
END ;
proc.dsc := topScope.next;
IF sym = semicolon THEN Get(sym) ELSE Mark(";?") END;
locblksize := 0; declarations(locblksize);
WHILE sym = procedure DO
    ProcedureDecl;
    IF sym = semicolon THEN Get(sym) ELSE Mark(";?") END
END ;
proc.val := pc; Enter(locblksize);
IF sym = begin THEN Get(sym); StatSequence END ;
IF sym = end THEN Get(sym) ELSE Mark("END?") END ;
IF sym = ident THEN
    IF procid # id THEN Mark("no match") END ;
    Get(sym)
ELSE Mark("ident?")
END ;
Return(parblksize - 8); CloseScope; DEC(level)
END
END ProcedureDecl;

```

Within the procedure body, value parameters are treated exactly like local variables. Their entries in the symbol table are of class *Var*. A new class *Par* is introduced to represent reference parameters. The addresses (offsets) of formal parameters are derived according to the following formula, whereby the last parameter *pn* obtains the least offset, namely the size of the procedure mark (8). The size of variable parameters is always 4, which is the size of an address.

$$\text{adr}(\text{pi}) = \text{size}(\text{pi}+1) + \dots + \text{size}(\text{pn}) + 8$$

Unfortunately, this implies that the offsets cannot be determined before the entire parameter list has been recognized. In the case of byte-addressed stores it is moreover advantageous always to increment or decrement the stack pointer by multiples of 4, such that parameters are always aligned to word boundaries. In the case of Oberon-0 special attention to this rule is unnecessary, because all data types feature a size of multiples of 4 anyway.

Local declarations are processed by the parser procedure *declarations*. The code for the prologue is emitted by procedure *Enter* after the processing of local declarations. Emission of the epilogue is performed by procedure *Return* at the end of *ProcedureDecl*.

```

PROCEDURE Enter(size: LONGINT);
BEGIN
    Put(PSH, LNK, SP, 4);
    Put(PSH, FP, SP, 4);
    Put(MOV, FP, 0, SP);
    Put(SUBI, SP, SP, size)
END Enter;

PROCEDURE Return(size: LONGINT);
BEGIN
    Put(MOV, SP, 0, FP);
    Put(POP, FP, SP, 4);
    Put(POP, LNK, SP, size+4);
    PutBR(RET, LNK)
END Return;

```


Procedure *MakeItem* generates an Item corresponding to a given *Object*. At this point, the difference between the addressing of local and global variables must be taken into account. (As already mentioned, the handling of intermediate-level variables is not treated here.) Note, however, that reference parameters (*class* = *Par*) require indirect addressing. Since the RISC architecture does not explicitly feature an indirect addressing mode, the value of the formal parameter, which is the address of the actual parameter, is loaded into a register. The actual parameter is then accessed via this register, with offset 0.

```

PROCEDURE MakeItem(VAR x: Item; y: Object);
  VAR r: LONGINT;
BEGIN x.mode := y.class; x.type := y.type; x.a := y.val;
  IF y.lev = 0 THEN x.r := PC ELSEIF y.lev = curlev THEN x.r := FP
  ELSE Mark("level!"); x.r := 0
  END ;
  IF y.class = Par THEN GetReg(r); Put(LDW, r, x.r, x.a); x.mode := Var; x.r := r; x.a := 0 END
END MakeItem;

```

Procedure calls are generated within the already encountered procedure *StatSequence* with the aid of auxiliary procedures *Parameter* and *Call*:

```

IF sym = ident THEN
  find(obj); Get(sym); MakeItem(x, obj); selector(x);
  IF sym = becomes THEN ...
  ELSIF x.mode = Proc THEN
    par := obj.dsc;
    IF sym = lparen THEN Get(sym);
    IF sym = rparen THEN Get(sym)
    ELSE
      LOOP expression(y);
        IF IsParam(par) THEN Parameter(y, par.type, par.class); par := par.next
        ELSE Mark("too many parameters")
        END ;
        IF sym = comma THEN Get(sym)
        ELSIF sym = rparen THEN Get(sym); EXIT
        ELSIF sym >= semicolon THEN Mark(" ?"); EXIT
        ELSE Mark(" or , ?")
        END
      END
    END
  END ;
  IF obj.val < 0 THEN Mark("forward call")
  ELSIF ~IsParam(par) THEN Call(x)
  ELSE Mark("too few parameters")
  END

```

...

```

PROCEDURE Parameter(VAR x: Item; ftyp: Type; class: INTEGER);
  VAR r: LONGINT;
BEGIN
  IF x.type = ftyp THEN
    IF class = Par THEN (*Var param*)
      IF x.mode = Var THEN
        IF x.a # 0 THEN GetReg(r); Put(ADDI, r, x.r, x.a) ELSE r := x.r END
        ELSE Mark("illegal parameter mode")
        END ;
        Put(PSH, r, SP, 4); EXCL(regs, r) (*push*)
      ELSE (*value param*)
        IF x.mode # Reg THEN load(x) END ;
        Put(PSH, x.r, SP, 4); EXCL(regs, x.r)
      END
    END
  END

```

```

    END
    ELSE Mark("bad parameter type")
    END
END Parameter;

PROCEDURE IsParam(obj: Object): BOOLEAN;
BEGIN RETURN (obj.class = Par) OR (obj.class = Var) & (obj.val > 0)
END IsParam;

PROCEDURE Call(VAR x: Item);
BEGIN PutBR(BSR, x.a - pc)
END Call;

```

Here we tacitly assume that the entry addresses of procedures are known when a call is to be compiled. Thereby we exclude forward references which may, for example, arise in the case of mutual, recursive referencing. If this restriction is to be lifted, the locations of forward calls must be retained in order that the branch instructions may be fixed up when their destinations become known. This case is similar to the fixups required for forward jumps in conditional and repeated statements.

In conclusion, we show the code generated for the following, simple procedure:

```

PROCEDURE P(x: INTEGER; VAR y: INTEGER);
BEGIN x := y; y := x; P(x, y); P(y, x)
END P

```

0	PSH	LNK, SP,	4	prologue
4	PSH	FP, SP,	4	
8	MOV	FP, 0,	SP	
12	SUBI	SP, SP,	0	no local variables
16	LDW	0, FP,	8	
20	LDW	1, 0,	0	
24	STW	1, FP,	12	x := y
28	LDW	0, FP,	8	
32	LDW	1, FP,	12	
36	STW	1, 0,	0	y := x
40	LDW	0, FP,	12	x
44	PSH	0, SP,	4	
48	LDW	0, FP,	8	adr(y)
52	PSH	0, SP,	4	
56	BSR		-14	P(x, y)
60	LDW	0, FP,	8	
64	LDW	1, 0,	0	y
68	PSH	1, SP,	4	
72	ADDI	0, FP,	12	adr(x)
76	PSH	0, SP,	4	
80	BSR		-20	P(y, x)
84	MOV	SP, 0,	FP	epilogue
88	POP	FP, SP,	4	
92	POP	LNK, SP,	12	pop link and parameters
96	RET	LNK		

12.5. Standard procedures

Most programming languages feature certain procedures and functions which do not need to be declared in a program. They are said to be *predeclared* and they can be called from anywhere, as they are pervasive. These are well-known functions, for example the absolute value of a number (ABS), type

conversions (ENTIER, ORD), or frequently encountered statements which merit an abbreviation and are available on many computers as single instructions (INC, DEC). The property common to all these so-called standard procedures is that they correspond directly either to a single instruction or to a short sequence of instructions. Therefore, these procedures are handled quite differently by compilers; no call is generated. Instead, the necessary instructions are emitted directly into the code. These procedures are therefore also called *in-line* procedures, a term that makes sense only if the underlying implementation technique is understood.

As a consequence it is advantageous to consider standard procedures as an object class of their own. Thereby the need for a special treatment of calls becomes immediately apparent. For Oberon-0 we postulate procedures *Read*, *Write*, *WriteHex*, and *WriteLn*, which on the one hand introduce elementary input and output facilities, and on the other hand serve to demonstrate the proposed handling of predeclared procedures. In this case, the term standard is admittedly misleading, whereas predeclared and *in_line* refer to the core of the subject matter. The corresponding entries in the symbol table are made when the compiler is initialized, namely in an outermost scope called *universe* which always remains open (see Appendix C). The new class attribute is denoted by *SProc*, and attribute *val* (*a* in the case of *Items*) identifies the concerned procedure.

```

IF sym = ident THEN
  find(obj); Get(sym); MakeItem(x, obj); selector(x);
  IF sym = becomes THEN ...
  ELSIF x.mode = Proc THEN ...
  ELSIF x.mode = SProc THEN
    IF obj.val <= 3 THEN param(y); TestInt(y) END ;
    IOCall(x, y)
  ...
PROCEDURE IOCall(VAR x, y: Item);
  VAR z: Item;
  BEGIN (*x.mode = SProc*)
    IF x.a = 1 THEN (*Read*)
      GetReg(z.r); z.mode := Reg; z.type := intType; Put(RD, z.r, 0, 0); Store(y, z)
    ELSIF x.a = 2 THEN (*Write*) load(y); Put(WRD, 0, 0, y.r); EXCL(regs, y.r)
    ELSIF x.a = 3 THEN (*WriteHex*) load(y); Put(WRH, 0, 0, y.r); EXCL(regs, y.r)
    ELSE (*WriteLn*) Put(WRL, 0, 0, 0)
    END
  END IOCall;

```

The final example shows a sequence of three statements and the resulting code:

```

Read(x); Write(x); WriteLn
4   READ   0, 0, 0
8   STW    0, 0, -4      x
12  LDW    0, 0, -4      x
16  WRD    0, 0, 0
32  WRL    0, 0, 0

```

12.6. Function procedures

A function procedure is a procedure whose identifier simultaneously denotes both an algorithm and its result. It is activated not by a call statement but by a factor of an expression. The call of a function procedure must therefore also take care of returning the function's result. The question therefore arises of which resources should be used.

If our primary goal is the generation of efficient code with the minimal number of memory accesses, then a register is the prime candidate for temporarily holding the function's result. If this solution is adopted, we must renounce the capability of defining functions with a structured result, because structured values cannot be held in a register.

If this restriction is considered as unacceptable, a place in the stack must be reserved to hold the structured result. Typically, it is added to the parameter area of the activation record. The function result is considered as an implicit result (variable) parameter. Correspondingly, the stack pointer is incremented before code for the first parameter is emitted.

At this point, all the concepts contained in the language Oberon-0 and implemented in its compiler have been presented. The compiler's listing is contained in full in Appendix C.

12.7. Exercises.

- 12.1. Improve the Oberon-0 compiler in such a way that the restriction that variables must be strictly local or entirely global can be lifted.
- 12.2. Add standard functions to the Oberon-0 compiler, generating inline code. Consider ABS, INC, DEC.
- 12.3. Replace the VAR parameter concept by the notion of an OUT parameter. An OUT parameter represents a local variable whose value is assigned to its corresponding actual parameter upon termination of the procedure. It constitutes the inverse of the value parameter, where the value of the actual parameter is assigned to the formal variable upon the start of the procedure.

13. Elementary Data Types

13.1. The types REAL and LONGREAL

As early as 1957 integers and real numbers were treated as distinct data types in Fortran. This was not only because different, internal representations were necessary, but because it was recognized that the programmer must be aware of when computations could be expected to be exact (namely for integers), and when only approximate. The fact that with real numbers only approximate results can be obtained, may be understood by considering that real numbers are represented by scaled integers with a fixed, finite number of digits. Their type is called REAL, and a real value x is represented by the pair of integers e and m as defined by the equation

$$x = B^{e-w} \times m \quad 1 \leq m < B$$

This form is called *floating-point* representation; e is said to be the exponent, m the mantissa. The base B and the bias w are fixed values for all REAL values, characterizing the chosen number representation. The two IEEE standards of floating-point representations feature the following values for B and w , and to the components e and m a bit s is added for the sign:

Type	B	w	Number of bits for e	Number of bits for m	Total
REAL	2	127	8	23	32
LONGREAL	2	1023	11	52	64

The exact forms of the two types, called REAL and LONGREAL in Oberon, are specified by the following formulas:

$$x = (-1)^s \times 2^{e-127} \times 1.m \quad x = (-1)^s \times 2^{e-1023} \times 1.m$$

The following examples show the floating-point representation of some selected numbers:

Decimal	s	e	1.m	Binary	Hexadecimal
1.0	0	127	1.0	0 01111111 000000000000000000000000	3F80 0000
0.5	0	126	1.0	0 01111110 000000000000000000000000	3F00 0000
2.0	0	128	1.0	0 10000000 000000000000000000000000	4000 0000
10.0	0	130	1.25	0 10000010 010000000000000000000000	4120 0000
0.1	0	123	1.6	0 01111011 10011001100110011001101	3DC CCCD
-1.5	1	127	1.5	1 01111111 100000000000000000000000	BFC0 0000

Two examples illustrate the case of LONGREAL:

1.0	0	1023	1.0	0 01111111111 00000000 ... 00000000	3FF0 0000 0000 0000
0.1	0	1019	1.6	0 01111111011 10011001 ... 10011010	3FB9 9999 9999 999A

This logarithmic form inherently excludes a value for 0. The value 0 must be treated as a special case, and it is represented by all bits being 0. With regard to numeric properties it constitutes a special case and a discontinuity. Furthermore, the IEEE standards postulate two additional special values: $e = 0$ (with $m \neq 0$) and $e = 255$ (resp. $e = 1023$) are considered as invalid results and they are called *NaN* (not a number).

Normally, the programmer does not have to worry about these specifications, and the compiler designer is not affected by them. The types REAL and LONGREAL constitute abstract data types usually integrated in the hardware which features a set of instructions adapted to the floating-point representation. If this set is complete, that is, it covers all basic numeric operations, the representation may be considered as hidden, since no further, programmed operations depend on it. In many computers, instructions for floating-point operands use a special set of registers. The reason behind this is that often separate coprocessors, so-called *floating-point units* (FPUs) are used which implement all floating-point instructions and contain this set of floating-point registers.

13.2. Compatibility between numeric data types

The values of all variables with numeric data type are numbers. Therefore there is no obvious reason not to declare them all as assignment compatible. But, as already outlined, numbers of different types are differently represented in terms of bit sequences within the computer. Hence, whenever a number of type *T0* is assigned to a variable of type *T1*, a representation conversion has to be performed which takes some finite time. The question then arises of whether this fact should remain hidden from the programmer in order to avoid distraction, or whether it should be made explicit because it affects the efficiency of the program. The latter choice is accomplished by declaring the various types as incompatible and by providing explicit, predefined conversion functions.

In any case, in order to be complete, a computer's set of instructions must also contain conversion instructions which convert integers into floating-point numbers and vice-versa. The same holds at the level of the programming language.

In Oberon different data types may be used within an arithmetic expression, which is then said to be a mixed expression. In this case, the compiler must insert hidden conversion instructions yielding the representation required for the specified arithmetic operation to be applied.

The definition of Oberon provides not only two, but an entire set of numeric data types. They are ordered in the sense that the larger type contains all values belonging to the smaller type. This notion is called *type inclusion*:

$$\text{SHORTINT} \subseteq \text{INTEGER} \subseteq \text{LONGINT} \subseteq \text{REAL} \subseteq \text{LONGREAL}$$

The result type of an expression is defined to be equal to the larger of the operands. This rule determines the conversion instructions to be inserted by the compiler.

For the assignment, certain relaxed type compatibility rules are also postulated, relaxed relative to the strict rule that the type of the destination variable must be identical to that of the source expression. Oberon postulates that the type of the variable must include the type of the expression. The following examples are therefore permitted:

```
VAR i, j: INTEGER; k: LONGINT; x, y: REAL; z: LONGREAL;

i := j;      INTEGER M INTEGER      (no conversion)
k := i;      INTEGER M LONGINT      (INTEGER to LONGINT)
z := x;      REAL M LONGREAL        (REAL to LONGREAL)
x := i;      INTEGER M REAL          (INTEGER to REAL)
```

Conversions between integer types are simple and efficient, because they consist of a sign extension only. Much more involved are conversions from integer to floating-point representation. They consist of a normalization of the mantissa such that $1 \leq m < 2$, and the packing of sign, exponent and mantissa into a single word. Typically, an appropriate instruction is provided for this task.

However, if the type of the expression is larger than the type of the destination variable, the assignment is sometimes impossible, as the assigned value may be outside the range specified by the variable's type. Therefore, a run-time range check is necessary before the conversion. This fact should be made visible in the source language by an explicit conversion function. Oberon features the following functions:

```
SHORT      INTEGER to SHORTINT
           LONGINT to INTEGER
           LONGREAL to REAL

ENTIER     REAL to LONGINT
           LONGREAL to LONGINT
```

ENTIER(x) yields the largest integer not greater than x.

Oberon's type conversion rules are simple and easily explained. Nevertheless they harbour a pitfall. In a multiplication of two factors of type INTEGER (or REAL) in some cases it is tempting to expect a result of type LONGINT (resp. LONGREAL). However, in the assignments

$$k := i*j + k$$

$$z := x*y + z$$

the product is, according to the given rules, of type INTEGER (resp. REAL), and it is converted to its long variant for the subsequent addition. Computation of the product with higher precision is achieved by forcing the conversion to occur before multiplication:

$$k := \text{LONG}(i) * \text{LONG}(j) + k$$

$$z := \text{LONG}(x) * \text{LONG}(y) + z$$

The simplicity of the compatibility rules is a significant advantage for the compiler designer. The principle of expression handling is clearly prescribed. Only the compilation procedures for expressions and terms, and the one for assignment have to be adjusted by introducing case discriminations. The larger the number of different numeric types, the more cases have to be distinguished.

13.3. The data type SET

The units of storage in computers consist of a small number of bits which are interpretable in different ways. They may represent integers with or without sign, floating-point numbers or logical data. The question about the way to introduce logical bit sequences in higher programming languages has been controversial for a long time. The proposal to introduce them as sets is due to C. A. R. Hoare (Hoare, 1972).

The proposal is attractive, because the set is a mathematically well-founded abstraction. It is appropriately represented in a computer by its *characteristic function* F . If x is a set of elements from the ordered base set M , $F(x)$ is the sequence of truth values b_i with the meaning " i is contained in x ". If we choose a word (consisting of N bits) to represent values of type SET, the base set consists of the integers $0, 1, \dots, N-1$. N is typically so small that the range of applications for the type SET is quite restricted. However, the basic set operations of intersection, union and difference are implementable extremely efficiently. Examples of sets represented by bit sequences with word length N are:

x		N-1	...	7	6	5	4	3	2	1	0
{0, 2, 4, 6, ...}	0	...	0	1	0	1	0	1	0	1	1
{0, 3, 6, ...}	0	...	0	1	0	0	1	0	0	1	1
{ }	0	...	0	0	0	0	0	0	0	0	0

Oberon's set operators are implemented by logical instructions available on every computer. This is possible due to the following properties of the characteristic function. Note that we use the Oberon notation for set operations, that is, $x+y$ for the union and $x*y$ for the intersection:

$\forall i ((i \in x+y) : (i \in x) \text{ OR } (i \in y))$	Union
$\forall i ((i \in x*y) : (i \in x) \& (i \in y))$	Intersection
$\forall i ((i \in x-y) : (i \in x) \& \sim(i \in y))$	Difference
$\forall i ((i \in x/y) : (i \in x) \neq (i \in y))$	Symmetric difference

Consequently, the OR instruction can be used for set union, AND for set intersection and XOR for the symmetric difference. The result is a very efficient implementation, because the operation is executed on all elements (bits) simultaneously (in parallel). Examples with the base set $\{0, 1, 2, 3\}$ are:

$\{0, 1\} + \{0, 2\} = \{0, 1, 2\}$	0011	OR	0101	=	0111
$\{0, 1\} * \{0, 2\} = \{0\}$	0011	&	0101	=	0001
$\{0, 1\} - \{0, 2\} = \{1\}$	0011	& ~	0101	=	0010
$\{0, 1\} / \{0, 2\} = \{1, 2\}$	0011	XOR	0101	=	0110

We conclude by showing the code representing the set expression $(a+b) * (c+d)$

```
LDW  1, 0, a
LDW  2, 0, b
OR   1, 1, 2
LDW  2, 0, c
LDW  3, 0, d
OR   2, 2, 3
AND  1, 1, 2
```

The membership test $i \text{ IN } x$ is implemented by a bit test. If such an instruction is not available, the bit sequence must be shifted appropriately with subsequent sign bit test.

The type SET is particularly useful if the base set includes the ordinal numbers of a character set (CHAR). Efficiency is in this case somewhat reduced, because 256 bits (32 bytes) are typically required to represent a set value. Even in 32-bit computers 8 logical instructions are required for the execution of a set operation.

13.4. Exercises

13.1 Extend the language Oberon-0 and its compiler by the data type REAL (and/or LONGREAL) with its arithmetic operators +, -, * and /. The RISC architecture must be extended accordingly by a set of floating-point instructions and a set of floating-point registers. Choose one of the following alternatives:

- a. The result type of an operation is always that of the operands. The types INTEGER and REAL cannot be mixed. However, there exist the two transfer functions ENTIER(x) and REAL(i).
- b. Operands of the types INTEGER and REAL (and LONGREAL) may be mixed in expressions.

Compare the complexities of the compilers in the two cases.

13.2. Extend the language Oberon-0 and its compiler by the data type SET with its operators + (union), * (intersection) and - (difference), and with the relation IN (membership). Furthermore, set constructors are introduced by the following additional syntax. As an option, expressions in set constructors may be confined to constants.

```
factor = number | set | ...
set = "{" [element {"," element}] "}".
element = expression [".." expression].
```

13.3. Extend the language Oberon-0 and its compiler by the data type CHAR with the functions ORD(ch) (ordinal number of ch in the character set) and CHR(k) (k-th character in the character set). A variable of type CHAR occupies a single byte in store.

14. Open Arrays, Pointers and Procedure Types

14.1. Open arrays

An *open array* is an array parameter whose length is unknown (open) at the time of compilation. Here we encounter for the first time a situation where the size of the required storage block is not given. The solution is relatively simple in the case of a reference parameter, because no storage has to be allocated anyway, and merely a reference to the actual array is passed to the called procedure.

However, in order to check index bounds when accessing elements of the open array parameter, the length must be known. Therefore, in addition to the array's address, its length is also passed on. In the case of a multidimensional, open array the length is also necessary to compute element addresses. Hence, the length of the array in every dimension is supplied. The unit consisting of array address and lengths is called an *array descriptor*. Consider the following example:

```
VAR a: ARRAY 10 OF ARRAY 20 OF INTEGER;

PROCEDURE P(VAR x: ARRAY OF ARRAY OF INTEGER);
BEGIN k := x[i]
END P;

P(a)
```

A descriptor with three entries is pushed onto the stack as parameter to P (s. Figure 14.1) and the corresponding code is as follows:

MVI	1, 0, 20	R1 := 20
PSH	1, 30, 4	push len, R14 = SP
ADDI	1, 0, 10	R1 := 10
PSH	1, 30, 4	push len
ADDI	1, 0, a	R1 := adr(a)
PSH	1, 30, 4	push adr
BSR	P	call




Figure 14.1. Array descriptor for open array.

If an open array parameter is passed by value, its value must be copied into its provided formal location just as in the case of a scalar value. This operation may, however, take considerable effort if the array is large. In the case of structured parameters, programmers should always use the VAR option, unless a copy is essential.

Certainly the code for the copy operation is better inserted after the prologue of the procedure rather than in the place of the call. Consequently, the code pattern for the call is the same for value and reference parameters, with the exception that for the former the copy operation is omitted from the prologue.

The formal location apparently does not hold the array, but instead the array descriptor, whose size is known. The space for the copy is allocated at the top of the stack, and the stack pointer is incremented (or decremented) by the array's size. In the case of multidimensional arrays, the size is computed (at run-time) as the product of the individual lengths and the element size.

Here SP is changed at run time by amounts which are unknown at compile time. Therefore it is impossible in the general case to operate with a single address register (SP); the frame pointer FP is indeed necessary.

14.2. Dynamic data structures and pointers

The two forms of data structures provided in Oberon are the array (all elements of the same type, *homogeneous* structure) and the record (*heterogeneous* structure). More complex structures must be programmed individually, that is, they must be generated during program execution. For this reason they are said to be *dynamic* structures. Thereby the structure's components are generated one by one; storage is allocated for components individually. They do not necessarily lie in contiguous locations in store. Relationships between components are expressed explicitly by *pointers*.

For the implementation of this concept a mechanism must be available for the allocation of storage at run time. In Oberon, it is represented by the standard procedure NEW(x). This allocates storage to a dynamic variable, and assigns the address of the allocated block to the pointer variable x. From this it follows that pointers are addresses. Access to a variable referenced by a pointer is necessarily indirect as in the case of VAR parameters. In fact, a VAR parameter represents a hidden pointer. Consider the following declarations:

```
TYPE T = POINTER TO TDesc;
      TDesc = RECORD x, y : LONGINT END;
VAR a, b : T;
```

The code for the assignment $a.x := b.y$ with access via pointers a and b becomes

```
LDW  1, FP, b      R1 := b
LDW  2, 1, y        R2 := b.y
LDW  3, FP, a        R3 := a
STW  2, 3, x        a.x := R2
```

The step from the referencing pointer variable to the referenced record variable is called dereferencing. In Oberon the explicit *dereferencing* operator is denoted by the symbol \uparrow . $a.x$ is evidently an abbreviation for the more explicit form $a \uparrow .x$. The implicit dereferencing operation is recognizable when the selector symbol (dot) is preceded not by a record but by a pointer variable.

Everyone who has written programs which heavily involve pointer handling knows how easily errors can be made with catastrophic consequences. To explain why, consider the following type declarations:

```
T0 = RECORD x, y : LONGINT END ;
T1 = RECORD x, y, z : LONGINT END;
```

Let a and b be pointer variables, and let a point to a record of type $T0$, b to a record of type $T1$. Then the designator $a.z$ denotes an undefined value of a non-existent variable, and $a.z := b.x$ stores a value to some undefined location, perhaps corrupting another variable allocated to this location.

This dangerous situation is elegantly eliminated by binding pointers to a data type. This permits the validation of pointer values at the time of compilation without loss of run-time efficiency. This brilliant idea is due to C. A. R. Hoare and was implemented for the first time in Algol W (Hoare, 1972). The type to which a pointer is bound is called its *base type*.

```
P0 = POINTER TO T0;
P1 = POINTER TO T1;
```

Now the compiler can check and guarantee that only pointer values can be assigned to a pointer variable p which points to a variable of the base type of p . The value NIL, pointing to no variable at all, is considered as belonging to all pointer types. Referring to the example above, now the designator $a.z$ is detected as incorrect, because z is not a field of the type $T0$ to which a is bound. If every pointer variable is initialized to NIL, it suffices to precede every access via a pointer with a test for the pointer value NIL. In this case, the pointer points to no variable, and any designator must be erroneous.

Such a test is indeed quite simple, but because of its frequency it reduces efficiency. The need for an explicit code pattern can be circumvented by (ab)using the storage protection mechanism available on many computers. In this case, the test does not properly check whether $a = \text{NIL}$, but rather whether $a.z$ is a valid, unprotected address. If as usual NIL is represented by the address 0, and if locations $0 \dots N-1$ are protected, mistaken references via NIL are caught only if their field offsets are less than N . Nevertheless, the method seems to be satisfactory in practice.

The introduction of pointers requires a new class of objects in the symbol table and also a new mode of items. Both are to imply indirect addressing. Because VAR parameters also require indirect addressing, a mode indicating indirection is already present, and it is only natural to use the same mode for access via pointers. However, the name *Ind* would now appear as more appropriate than *Par*.

Designator	Mode	
x	Var	Direct addressing
x^\wedge	Ind	Indirect addressing
$x^\wedge.y$	Ind	Indirect addressing with offset

Hence, the (usually implied) dereferencing operator converts the mode of an item from *Var* to *Ind*. To summarize:

1. The notion of a pointer is easily integrated into our system of type compatibility checking. Every pointer type is bound to a base type, namely the type of the referenced variable.
2. x^\wedge denotes dereferencing, implemented by indirect addressing.
3. Pointers are type safe if access is preceded by a NIL test, and if pointer variables are initialized to NIL .

Allocation of variables referenced via pointers is obtained by a call of procedure $\text{NEW}(p)$. We postulate its existence as run-time support in operating systems. The size of the block to be allocated is given by the base type of p .

So far, we have ignored the problem of storage reclamation. It is actually irrelevant for abstract programs; for concrete ones, however, it is crucial, as stores are inherently finite. Modern operating systems offer a centralized storage management with garbage collection. There are various schemes for storage reclamation; but we shall not explain them here. We restrict ourselves to the only question relevant to the compiler designer: which data must be provided to the garbage collector, so that at any time all irrelevant storage blocks can safely be identified and reclaimed? A variable is no longer relevant when there are no references to it, references emanating from declared pointer variables. In order to determine whether such references exist, the garbage collector requires the following data:

1. the addresses of all declared pointer variables,
2. the offsets of all pointer fields in dynamically allocated records, and
3. the size of every dynamically allocated variable.

This information is available at compile time, and it has to be "handed down" in such a way that it is available to the garbage collector at run time. In this sense compiler and system must be integrated. The system is here assumed to include storage management, in particular the allocator NEW and the garbage collector.

In order to make this information available at run time, procedure NEW not only allocates a block of storage, but provides it with a type description of the allocated variable. Naturally, such a descriptor must be issued only once, as it need not be duplicated for every instance (variable) of the same type. Therefore, the block is assigned merely a pointer to the type descriptor, and this pointer remains invisible to the programmer. The pointer is called a type tag (s. Figure 14.2).

The type descriptor apparently is a reduced form of the object describing the type in the compiler's symbol table, reduced to the data relevant for storage reclamation. This concept has the following consequences:

1. The compiler must generate a descriptor for every (record) type, and it must add it to the object file.

2. Procedure NEW(p) obtains, in addition to the address of p, an additional, hidden parameter specifying the address of the descriptor of the base type of p.
3. The program loader must interpret the added object file information and generate type descriptors.




Figure 14.2. Pointer variable, referenced variable, and type descriptor.

The type descriptor specifies the size of the variable and the offset of all pointer fields (Figure 14.3).




Figure 14.3. Variable with type descriptor.

This, however, is still insufficient. In order that data structures can be traversed, their roots have to be known. Therefore, the object file is also provided with a list of all declared pointer variables. This list is copied upon loading into memory. The list must also include the hidden pointers designating type descriptors. In order that descriptors do not have to be generated for all data types, Oberon restricts pointers to refer to records. This is justified when considering the role of records in dynamic data structures.

14.3. Procedure types

If in a language procedures can be passed as parameters, or if they can occur as values of variables, it becomes necessary to introduce *procedure types*. Which are the characteristics of such types, that is, of the values which variables may assume?

Procedure types have been in use since the advent of Algol 60. There, they occurred implicitly only. A parameter in Algol 60 can be a procedure (formal procedure). Its type, however, is not specified; it is merely known that the parameter denotes some procedure or function. The type specification is incomplete or missing, and this constitutes an unfortunate loophole in Algol's type system. In Pascal, it was retained as a concession to Algol compatibility. Modula-2, however, requires a complete, type-safe specification, and besides parameters, variables with procedures as their values are also allowed. Thereby procedure types achieve the same standing as other data types. In this respect, Oberon has adopted the same concept as Modula-2 (Wirth, 1982).

What does this type-safe specification, called the procedure's *signature*, consist of? It contains all specifications necessary to validate the compatibility between actual and formal parameters, namely their number, the type of each parameter, its kind (value or reference) and the type of the result in the case of function procedures. The following example illustrates the case:

```

PROCEDURE F(x, y : REAL): REAL;
BEGIN
...
END F

PROCEDURE H(f: PROCEDURE (u, v : REAL): REAL);
  VAR a, b: REAL;
  BEGIN a := f(a + b, a - b)
  END H

```

Upon compilation of the declaration of H the type compatibility between $a + b$ and u , respectively that between $a - b$ and v , is checked, as well as whether the result type of f is assignable to a . In the call $H(F)$ the compatibility between the parameters, and that of the result type of the actual F and the formal f is verified, that is, between x and u and between y and v . Note that the identifiers u and v do not occur in the program, except as the names of the formal parameters of the formal procedure f . Hence, they are actually superfluous, but they may be useful as comments to the reader if meaningful names are chosen.

Pascal, Modula and Oberon assume *name compatibility* as the basis for establishing type consistency. In the case of procedure parameters, an exception was made; *structural compatibility* suffices. If name compatibility were required, the type (signature) of every procedure used as an actual parameter would have to be given an explicit name. This was considered as too cumbersome when the language was designed. However, structural compatibility requires that a compiler be capable of comparing two parameter lists for type correspondence.

A procedure may thus be assigned to a variable under the condition that the two parameter lists correspond. The assigned procedure is activated by referring to the procedure variable. The call is indirect. This is actually the basis of object-oriented programming, where procedures are bound to fields of record variables called *objects*. These bound procedures are called methods. In contrast to Oberon, methods, once declared and bound, cannot be altered. All instances of a class refer to the same methods.

The implementation of procedure types and methods turns out to be surprisingly simple, if the problem of type compatibility checking is ignored. The value of a variable or record field with procedure type is simply the entry address of the assigned procedure. This holds only if we require that only global procedures, that is, procedures which are not embedded in some context, can be assigned. This readily acceptable restriction is explained with the aid of the following example which breaches this restriction. Upon execution of Q alias v the context containing variables a and b is missing.

```

TYPE T = PROCEDURE (u: INTEGER);
VAR v: T; r: INTEGER;

PROCEDURE P;
  VAR a, b: INTEGER;
  PROCEDURE Q(VAR x: INTEGER);
    BEGIN x := a+b END Q;
  BEGIN v := Q
  END P;

... P; v(r) ...

```

14.4. Exercises.

14.1. Extend the language Oberon-0 and its compiler with open arrays:

- a. for one-dimensional VAR parameters,
- b. for multi-dimensional VAR parameters,
- c. for value parameters.

14.2. Extend the language Oberon-0 and its compiler with function procedures:

- a. for scalar result types (INTEGER, REAL, SET),
- b. for any type.

14.3. A certain module M manages a data structure whose details are to be kept hidden. In spite of this hiding it must be possible to apply any given operation P on all elements of the structure. For this purpose, a procedure *Enumerate* is exported from M , which allows P to be specified as parameter. As a simple example, we choose for P the counting of the elements currently in the data structure and display the desired solution:

```
PROCEDURE Enumerate(P: PROCEDURE (e: Element));
  PROCEDURE CountElements*;
    VAR n: INTEGER;
    PROCEDURE Cnt(e: Element); BEGIN n := n+1 END Cnt;
  BEGIN n := 0; M.Enumerate(Cnt); Texts.WriteInt(W, n, 6)
  END CountElements;
```

Unfortunately, this solution violates a restriction postulated for the language Oberon. The restriction specifies that procedures used as parameters must be declared globally. This forces us to declare *Cnt* outside of *CountElements* and thereby also the counter variable n , although both definitely have no global function.

Implement procedure types in such a way that the mentioned restriction can be lifted, and that the proposed solution is admissible. What is the price?

14.4. Our RISC (see Ch. 9) features a push and a pop instruction. They are used to implement the stack paradigm for procedure activation frames. If we add an alternative push instruction PSHu, we can use it, together with the existing pop instruction, to implement fifo buffers:

```
PSHu: M[R[b] div 4] := R[a]; INC(R[b], c)
```

But how is the concept of buffer and the use of the buffer instructions to be represented in the language Oberon-0? We suggest the concept of the *rider*. Suppose that a rider r and a buffer B are declared as

```
VAR B: ARRAY N OF INTEGER;
    r: RIDER ON B;
```

and that riders are *allocated in registers*. Then let the following operators be defined on riders:

```
SET(r, B, i)    Let rider  $r$  point at  $B[i]$ 
```

r^{\wedge} designates the element pointed at by r . After access or assignment r is advanced to the next element of the buffer.

Write an appropriate module with a buffer and procedures for fetching and storing elements. Investigate the advantages of this concept, and also its problems. Implement it by additions to the compiler.

14.5. Our RISC (see Ch. 9) is said to be *byte-oriented*. But data are always moved to and from memory in groups of 4 bytes, called *words*. Word addresses are always multiples of 4. The least two bits of addresses are ignored.

Let us make use of byte addressing by extending the concept of riders (see 14.4) to buffers of bytes (characters). We extend the RISC's instruction set by the two instructions PSHB and POPB. They use a register as a 4-byte buffer, and transfer a word from of to memoey only after every fourth access. The pop instruction is defined as follows:

```
POPB: rotate R[a] by 8 bits to the right;
      IF R[b] MOD 4 = 0 THEN R[a] := M[R.b DIV 4] END ;
      R[b] := R[b] + 1
```

The states of $R[a]$ and $R[b]$ before and after every POPB instruction are shown below. After each instruction, the next byte is available in the least byte of register $R[a]$:

$R[a]$	$R[b]$	POPB	$R[a]$	$R[b]$
	0	load word	DCBA	1
DCBA	1	rotate	ADCB	2

ADCB	2	rotate	BADC	3
BADC	3	rotate	CBAD	0
CBAD	0	load word	...	1

Similarly, the push instruction is defined as follows:

```

PSHB: rotate R[a] by 8 bits to the right;
      IF R[b] MOD 4 = 3 THEN M[R.b DIV 4] := R[a] END ;
      R[b] := R[b] + 1

```

Before every push instruction, the next byte to be stored is placed in the least byte of the buffer register R[a]. The states of R[a] and R[b] before and after every POPB instruction are shown below:

R[a]	R[b]	PSHB	R[a]	R[b]
... A	0	rotate	A...	1
A... B	1	rotate	BA...	2
BA... C	2	rotate	CBA...	3
CBAD	3	store rotated word	DCBA	0

Define the *byte rider* construct in the language Oberon-0 and implement it by extending the compiler.

15. Modules and Separate Compilation

15.1. The principle of information hiding

Algol 60 introduced the principles of textual locality of identifiers and that of limited lifetime of the identified objects during execution. The range of visibility of an identifier in the text is called scope (Naur, 1960), and it extends over the block in which the identifier is declared. According to the syntax, blocks may be nested, with the consequence that the rules about visibility and scopes must be refined. Algol 60 postulates that identifiers declared in a block B are visible within B and within all blocks contained in B. But they are invisible in the environment of B.

From this rule, the implementer concludes that storage must be allocated to a variable *x* local to B as soon as control enters B, and that storage may be released as soon as control leaves B. Not only is *x* invisible outside B, but *x* ceases to exist when control is outside B. This implies the significant advantage that storage need not remain allocated to all variables of a program.

In some cases, however, the continued existence of a variable during a period of invisibility is highly desirable. Variable *x* then seems to reappear with its previous value as soon as control reenters block B. This special case was covered in Algol 60 by the feature of own variables. But this solution was soon discovered to be quite unsatisfactory, in particular in connection with recursive procedures.

An elegant and highly useful solution to the own-problem was discovered around 1972 with the structure of the module. It was adopted in the languages Modula (Wirth, 1977) and Mesa (Mitchell, Maybury and Sweet, 1978), and later under the name package in Ada. Syntactically, a module resembles a procedure and consists of local declarations followed by statements. In contrast to a procedure, however, a module is not called, but its statements are executed once only, namely when the module is loaded. The locally declared objects are static and remain in existence as long as the module remains loaded. The statements in the module body merely serve to initialize the module's variables. These variables are invisible outside the module; effectively they are hidden. D. L. Parnas has coined the term information hiding, and it has become an important notion in software construction. Oberon features the possibility of specifying selected identifiers declared in modules as visible in the module's environment. These identifiers are then said to be exported.

The own variable *x* declared within the Algol procedure *P* now will be declared, like *P* itself, local to a module *M*. *P* is exported, but not *x*. In the environment of *M* the details of the implementation of *P* as well as the variable *x* are hidden, but *x* retains its existence and its value between calls of *P*.

The desire to hide certain objects and details is particularly justified if a system consist of various parts whose tasks are relatively well separated, and if the parts themselves are of a certain complexity. This is typically the case in an organizational unit which manages a data structure. Then the data structure is hidden within a module, and it is accessible only via exported procedures. The programmer of this module postulates certain invariants, such as consistency conditions, which govern the data structure. These invariants can be guaranteed to hold, because they cannot be violated by parts of the system outside the module. As a consequence, the programmer's responsibility is effectively limited to the procedures within the module. This encapsulation of details solely responsible for the specified invariants is the true purpose of information hiding and of the module concept.

Typical examples of modules and information hiding are the file system hiding the structure of files and their dictionary, the scanner of a compiler hiding the source text and its lexicographic structure, or the code generator of a compiler hiding the generated code and the structure of the target architecture.

15.2. Separate compilation

It is tempting to postulate that modules be nestable like procedures. This facility is offered for example by the language Modula-2. In practice, however, this flexibility has hardly been fruitful. A flat module structure usually suffices. Hence, we consider all modules as being global, and their environment as the universe.

Much more relevant than their nestability is the possibility of developing and compiling modules separately. The latter is clearly feasible only if the modules are global, that is, not nested. The reason for this demand is simply the fact that software is never planned, implemented and tested in straight sequence, but that it is developed in steps, each step incorporating some additions or adaptations. Software is not "written", but grows. The module concept is of fundamental importance in this connection, because it allows development of individual modules separately under the assumption of constant interfaces of their imports. The set of exported objects effectively constitutes a module's interface with its partners. If an interface remains unchanged, a module's implementation can be improved (and corrected) without needing to adapt and recompile the module's clients. This is the real justification for separate compilation.

The advantage of this concept becomes particularly relevant if software is developed by teams. Once agreement is reached about the partitioning of a system into modules and about their interfaces, the team members can proceed independently in implementing the module assigned to them. Even if in practice it turns out that later changes in the specification of interfaces are avoidable only rarely, the simplification of teamwork through the concept of separate compilation of modules can hardly be overestimated. The successful development of complex systems crucially depends on the concept of modules and their separate compilation.

At this point, the reader may think that all this is not new, that the independent programming of modules and their binding by the program loader, as symbolized in Figure 15.1, has been in common use since the era of assemblers and the first Fortran compilers.




Figure 15.1. Independent compilation of modules A and B.

However, this ignores the fact that higher programming languages offer significantly increased protection against mistakes and inconsistencies through their static type concept. This inestimable - but all too often underestimated gain - is swept aside if type consistency checks are guaranteed only within modules, but not across module boundaries. This implies that type information about all imported objects must be available whenever a module is compiled. In contrast to independent compilation (Figure 15.1), where this information is not available, compilation as shown in Figure 15.2 with type consistency checks across module boundaries is called separate compilation.

Information about the imported objects is essentially an excerpt of the symbol table as presented in Chapter 8. This excerpt of the symbol table, transformed into a sequential form, is called a symbol file. Compilation of a module A which imports (objects from) modules B1 ... Bn now requires, in addition to the source text of A, the symbol files of B1 ... Bn. And in addition to the object code (A.obj) it also generates a symbol file (A.sym).




Figure 15.2. Separate compilation of modules A and B.

15.3. Implementation of symbol files

From the foregoing considerations we may first conclude that compilation of a module's import list causes a symbol file to be read for each module identifier in the list. The symbol table of the compiled module is initialized by the imported symbol files. Second, it follows that at the end of compilation the new symbol table is traversed, and a symbol file is output with an entry corresponding to every symbol table element marked for export. Figure 15.3 shows as an example the relevant excerpt of the symbol table during compilation of a module A importing B. Within B, T and f are marked with an asterisk for export.




Figure 15.3. Symbol table of A with imports from B.

Let us first consider the generation of the symbol file M.sym of a module M. At first sight, the task merely consists of traversing the table and emitting an entry corresponding to every marked element in an appropriately sequentialized form. The symbol table is essentially a list of objects with pointers to type structures which are trees. In this case the sequentialization of structures using a characteristic prefix for every element is perhaps the most appropriate technique. It is illustrated by an example in Figure 15.4.

```
VAR x: ARRAY 10 OF INTEGER;
    b: ARRAY 8 OF ARRAY 20 OF REAL
```




Figure 15.4. Sequentialized form of a symbol table with two arrays.

A problem arises because every object contains at least a pointer referring to its type. Writing pointer values into a file is problematic, to say the least. Our solution consists in writing the type description into the file the first time it is encountered when scanning the symbol table. Thereby the type entry is marked and obtains a unique reference number. The number is stored in an additional record field of the type ObjectDesc. If the type is referenced again later, the reference number is output instead of the structure.

This technique not only avoids the repeated writing of the same type descriptions, but also solves the problem of recursive references, as shown in Figure 15.5.




Figure 15.5. Cyclic reference of type node.

Positive values are used for reference numbers. As an indication that the reference number is used for the first time, and that it is therefore immediately followed by the type description, the number is given a negative sign. While reading a symbol file, a type table T is constructed with references to the respective type structures. If a positive reference number r is read, T[r] is the needed pointer; if r is negative, the subsequent type data is read, and the pointer referring to the newly constructed descriptor is assigned to T[-r].




Figure 15.6. Re_export of type A.T from module B.

Type information can, in contrast to data about other objects, be imported and at the same time be re-exported. Therefore it is necessary to specify the module from which the exported type stems. In order to make this possible, we use a so-called module anchor. In the heading of every symbol file there is a list

of anchor objects, one for each imported module which is re_exported, that is, which contains a type that is referenced by an exported object. Figure 15.6 illustrates such a situation; module C imports modules A and B, whereby a variable x is imported from B whose type stems from A. The type compatibility check for an assignment like $y := x$ rests on the assumption that the type pointers of x and y both refer to the same type descriptor. If they do not, an error is indicated.

Hence we conclude that upon compilation of a module M, not only the symbol tables of the explicitly imported modules must be present, but also those of modules from which types are referenced either directly or indirectly. This is a cause for concern, because the compilation of any module might necessitate the reading of symbol files of entire module hierarchies. It might even reach down to the deepest level of an operating environment, from where neither variables nor procedures are imported, but perhaps only a single type. The result would not only be the superfluous loading of large amounts of data, but also a waste of much memory space. It turns out, that although our concern is justified, the consequences are much less dramatic than might be expected (Franz, 1993). The reason is that most symbol tables requested are present already for other reasons. As a consequence, the additional effort remains small. Nevertheless it is worth pondering over the possibility of avoiding the extra effort. Indeed, the first compilers for Modula and Oberon have adopted the following technique.

Let a module M import types from modules M0, M1, and so on, either directly or indirectly. The solution consists of including in the symbol file of M complete descriptions of the imported types, thereby avoiding references to the secondary modules M0, M1, and so on. However, this fairly obvious solution causes complication. In the example illustrated by Figure 15.6, the symbol file of B evidently contains a complete description of type T. The consistency check for the assignment $y := B.x$, in order to be highly efficient, merely compares two type pointers. The configuration shown on the right of Figure 15.6 must therefore be present after loading. This implies that in symbol files re-exported types not only specify their home module, but that when loading a symbol file a test must verify whether or not the read type is already present. This may be the case because the symbol file of the module defining the type has already been loaded, or because the type has already been read when loading other symbol files.

At this point we also mention another, small complication in connection with types that arises because types may appear under different names (aliases). Although use of aliases is rare, the language definition (unfortunately) allows it. They are moderately meaningful only if the synonyms stem from different modules, as shown in Figure 15.7.




Figure 15.7. Type with aliases.

When loading the symbol file of B it is recognized that B.T1 and A.T0, both pointing to a type object, must actually point to the same object descriptor. In order to determine which of the two descriptors should be discarded and which one retained, type nodes (type Structure) are supplied with a back-pointer to the original type object (type Object), here to T0.

15.4. Addressing external objects

The principal advantage of separate compilation is that changes in a module *M* do not invalidate clients of *M*, if the interface of *M* remains unaffected. We recall that the interface consists of the entire set of exported declarations. Changes which do not affect the interface may occur, so to say, under cover, and without client programmers being aware of them. Such changes must not even require recompilation of the clients using new symbol files. For the sake of honesty, we hasten to add that exported procedures must in their semantics not have altered, because compilers could not detect such changes reliably. Hence, if we say that an interface remains unchanged, we explicitly refer to the declarations of types and variables, and to the signatures of procedures, and only implicitly to their semantics.

If in a certain module non-exported procedures and variables are changed, added or deleted, their addresses necessarily also change, and as a consequence so do those of other, possibly exported variables and procedures. This leads to a change of the symbol table, and thereby also to an invalidation of client modules. But this obviously contradicts the requirements postulated for separate compilation.

The solution to this dilemma lies in avoiding the inclusion of addresses in a symbol file. This has the consequence that addresses must be computed when loading and binding a module. Hence, in addition to its address (for module-internal use), an exported object is given a unique number. This number assumes the place of the address in the symbol file. Typically, these numbers are allocated strictly sequentially.

As a consequence, when compiling a client, only module specific numbers are available, but no addresses. These numbers must, as mentioned before, be converted into absolute addresses upon loading. For this task, knowledge about the positions of such incomplete address fields must be available. Instead of supplying the object file with a list of all locations of such addresses, the elements of this fixup list are embedded in the instructions at the very places of the yet unknown addresses. This mirrors the technique used for the completion of addresses of forward jumps (see Chapter 11). If all such addresses to be completed are collected in a single fixup list, then this corresponds to the Figure 15.8 (a). Every element must be identified with a pair consisting of a module number (*mno*) and an entry number (*eno*). It is simpler to provide a separate list for every module. In the object file, not just a single fixup root, but one for each list is required. This corresponds to the Figure 15.8 (b). Part (c) shows the extreme solution where a separate fixup list is specified for every imported object. Which of the three presented solutions is adopted, depends on how much information can be put into the place of an absolute address, by which it is ultimately replaced.




Figure 15.8. Three forms of fixup lists in object files.

15.5. Checking configuration consistency

It may seem belated if we now pose the question: Why are symbol files introduced at all? Let us assume that a module M is to be compiled which imports M0 and M1. A rather straightforward solution would be to recompile M0 and M1 immediately preceding the compilation of M, and to unite the three symbol tables obtained. The compilations of M0 and M1 might easily be triggered by the compilation of M reading the import list.

Although the repeated compilation of the same source text is a waste of time, this technique is used by various commercial compilers for (extended) Pascal and Modula. The serious shortcoming inherent in this method, however, is not so much the additional effort needed, but the lack of a guarantee for the consistency of the modules being bound. Let us assume that M is to be recompiled after some changes had been made in the source text. Then it is quite likely that after the original formulation of M and after its compilation, changes have also been made to M0 and M1. These changes may invalidate M. Perhaps even the source versions of M0 and M1 currently available to the programmer of client M no longer comply with the actual object files of M0 and M1. This fact, however, cannot be determined by a compilation of M, but it almost certainly leads to disaster when the inconsistent parts are bound and executed.

Symbol files, however, do not permit changes like source files; they are encoded and not visible through a text editor. They can only be replaced as a whole. In order to establish consistency, every symbol file is provided with a unique key. Symbol files thus make it possible to make modules available without giving away the source text. A client may rely on the specified interface definition and, thanks to the key, the consistency of the definition with the present implementations is also guaranteed. Unless this guarantee is provided, the entire notion of modules and separate compilation is perhaps enticing, but hardly a useful tool.




Figure 15.9. Inconsistency of module versions.

As an example, Figure 15.9 shows on its left side the situation upon compilation of module B, and on the right side that upon compilation of module C. Between the two compilations, A was changed and recompiled. The symbol files of B and C therefore contain module anchors of A with differing keys, namely 8325 in B and 8912 in C. The compiler checks the keys, notices the difference, and issues an error message. If, however, module A is changed after the recompilation of C (with changed interface), then the inconsistency can and must be detected upon loading and binding the module configuration. For this purpose, the same keys are also included in the respective object files. Therefore it is possible to detect the inconsistency of the import of A in B and C before execution is attempted. This is absolutely essential.

The key and the name are taken as the characteristic pair of every module, and this pair is contained in the heading of every symbol and object file. As already mentioned, the names of modules in the import list are also supplemented by their key. These considerations lead to the structure of symbol files as specified in Appendix A.3.

Unique module keys can be generated by various algorithms. The simplest is perhaps the use of current time and date which, suitably encoded, yield the desired key. A drawback is that this method is not entirely reliable. Even if the resolution of the clock is one second, simultaneous compilations on different

computers may generate the same key. Somewhat more significant is the argument that two compilations of the same source text should always generate the same key; but they do not. Hence, if a change is made in a module which is later detected to be in error, recompilation of the original version nevertheless results in a new key which lets old clients appear as invalidated.

A better method to generate a key is to use the symbol file itself as argument, like in the computation of a checksum. But this method is also not entirely safe, because different symbol files may result in the same key. But it features the advantage that every recompilation of the same text generates the same key. Keys computed in this way are called fingerprints.

15.6. Exercises

15.1. Incorporate separate compilation into your Oberon-0 compiler. The language is extended to include an import list (see Appendix A.2) and a marker in the exported identifier's declaration. Use the technique of symbol files and introduce the rule that exported variables may not be assigned values from outside, that is, in importing modules that are considered to be read-only variables.

15.2. Implement a fingerprint facility for generating module keys.

16. Optimizations and the Frontend/Backend Structure

16.1. General considerations

If we analyse the code generated by the compiler developed in the preceding chapters, we can easily see that it is correct and fairly straightforward, but in many instances also improvable. The reason primarily lies in the directness of the chosen algorithm which translates language constructs independently of their context into fixed patterns of instruction sequences. It hardly perceives special cases and does not take advantage of them. The directness of this scheme leads to results that are only partially satisfactory as far as economy of storage and execution speed are concerned. This is not surprising, as source and target languages do not correspond in simple ways. In this connection we can observe the semantic gap between programming language on the one hand and instruction set and machine architecture on the other.

In order to generate code which utilizes the available instructions and machine resources more effectively, more sophisticated translation schemes must be employed. They are called optimizations, and compilers using them are said to be optimizing compilers. It must be pointed out that this term, although in widespread use, basically is a euphemism. Nobody would be willing to claim that the code generated by them could be optimal in all cases, that is, in no way improvable. The so-called optimizations are nothing more than improvements. However, we shall comply with the common vocabulary and will also use the term optimization.

It is fairly evident that the more sophisticated the algorithm, the better the code obtained. In general it can be claimed that the better the generated code and the faster its execution, the more complex, larger and slower will be the compiler. In some cases, compilers have been built which allow a choice of an optimization level: while a program is under development, a low, and after its completion a high, degree of optimization is selected for compilation. As an aside, note that optimization may be selected with different goals, such as towards faster execution or towards denser code. The two criteria usually require different code generation algorithms and are often contradictory, a clear indication that there is no such thing as a well-defined optimum.

It is hardly surprising that certain measures for code improvement may yield considerable gains with modest effort, whereas others may require large increases in compiler complexity and size while yielding only moderate code improvements, simply because they apply in rare cases only. Indeed, there are tremendous differences in the ratio of effort to gain. Before the compiler designer decides to incorporate sophisticated optimization facilities, or before deciding to purchase a highly optimizing, slow and expensive compiler, it is worth while clarifying this ratio, and whether the promised improvements are truly needed.

Furthermore, we must distinguish between optimizations whose effects could also be obtained by a more appropriate formulation of the source program, and those where this is impossible. The first kind of optimization mainly serves the untalented or sloppy programmer, but merely burdens all the other users through the increased size and decreased speed of the compiler. As an extreme example, consider the case of a compiler which eliminates a multiplication if one factor has the value 1. The situation is completely different for the computation of the address of an array element, where the index must be multiplied by the size of the elements. Here, the case of a size equal to 1 is frequent, and the multiplication cannot be eliminated by a clever trick in the source program.

A further criterion in the classification of optimization facilities is whether or not they depend on a given target architecture. There are measures which can be explained solely in terms of the source language, independent of any target. Examples of target-independent optimizations are suggested by the following well known identities:

```
x + 0 = x
x * 2 = x + x
b & TRUE = b
b & ~b = FALSE
```



```
IF TRUE THEN A ELSE B END = A
IF FALSE THEN A ELSE B END = B
```

On the other hand, there are optimizations that are justified only through the properties of a given architecture. For example, computers exist which combine a multiplication and an addition, or an addition, a comparison and a conditional branch in a single instruction. A compiler must then recognize the code pattern which allows the use of such a special instruction.

Lastly, we must also point out that the more optimizations with sizeable effects that can be incorporated in a compiler, the poorer its original version must have been. In this connection, the cumbersome structures of many commercial compilers, whose origin is difficult to fathom, lead to surprisingly poor initial performance, which makes optimizing features seem absolutely indispensable.

16.2. Simple optimizations

First, let us consider optimizations that are implementable with modest effort, and which therefore are practically mandatory. This category includes the cases which can be recognized by inspection of the immediate context. A prime example is the evaluation of expressions with constants. This is called constant folding and is already contained in the compiler presented.

Another example is multiplication by a power of 2, which can be replaced by a simple, efficient shift instruction. Also, this case can be recognized without considering any context:

```
IF (y.mode = Const) & (y.a # 0) THEN
  n := y.a; k := 0;
  WHILE ~ODD(n) DO n := n DIV 2; k := k+1 END ;
  IF n = 1 THEN PutShift(x, k) ELSE PutOp(MUL, x, y) END
ELSE ...
END
```

Division (of integers) is treated in the same way. If the divisor is 2^k for some integer k , the dividend is merely shifted k bits to the right. For the modulo operator, the least significant k bits are simply masked out.

16.3. Avoiding repeated evaluation

Perhaps the best known case among the target independent optimizations is the elimination of common subexpressions. At first sight, this case may be classified among the elective optimizations, because the reevaluation of the same subexpression can be achieved by a simple change of the source program. For example, the assignments

```
x := (a+b)/c; y := (a+b)/d
```

can easily be replaced by three simpler assignments when using an auxiliary variable u :

```
u := a+b; x := u/c; y := u/d
```

Certainly, this is an optimization with respect to the number of arithmetic operations, but not with respect to the number of assignments or the clarity of the source text. Therefore the question remains open as to whether this change constitutes an improvement at all.

More critical is the case where the improvement is impossible to achieve by a change of the source text, as is shown in the following example:

```
a[i, j] := a[i, j] + b[i, j]
```

Here, the same address computation is performed three times, and each time it involves at least one multiplication and one addition. The common subexpressions are implicit and not directly visible in the source. An optimization can be performed only by the compiler.

Elimination of common expressions is only worth while if they are evaluated repeatedly. This may even be the case if the expression occurs only once in the source:

WHILE $i > 0$ DO $z := x+y$; $i := i-1$ END

Since x and y remain unchanged during the repetition, the sum need be computed only once. The compiler must pull the assignment to z out of the loop. The technical term for this feat is loop invariant code motion.

In all the latter cases code can only be improved by selective analysis of context. But this is precisely what increases the effort during compilation very significantly. The compiler presented for Oberon-0 does not constitute a suitable basis for this kind of optimization.

Related to the pulling out of constant expressions from loops is the technique of simplifying expressions by using the values computed in the previous repetition, that is, by considering recurrence relations. If, for example, the address of an array element is given by $\text{adr}(a[i]) = k*i + a_0$, then $\text{adr}(a[i+1]) = \text{adr}(a[i]) + k$. This case is particularly frequent and therefore relevant. For instance, the addresses of the indexed variables in the statement

FOR $i := 0$ TO $N-1$ DO $a[i] := b[i] * c[i]$ END

can be computed by a simple addition of a constant to their previous values. This optimization leads to significant reductions in computation time. A test with the following example of a matrix multiplication showed surprising results:

```
FOR i := 0 TO 99 DO
  FOR j := 0 TO 99 DO
    FOR k := 0 TO 99 DO  $a[i, j] := a[i, j] + b[i, k] * c[k, j]$  END
  END
END
```

The use of registers instead of memory locations to hold index values and sums, and the elimination of index bound tests resulted in a speed increased by a factor of 1.5. The replacement of indexed addressing by linear progression of addresses as described above yielded a factor of 2.75. And the additional use of a combined multiplication and addition instruction to compute the scalar products increased the factor to 3.90.

Unfortunately, not even consideration of simple context information suffices in this case. A sophisticated control and data flow analysis is required, as well as detection of the fact that in each repetition an index is incremented monotonically by 1.

16.4. Register allocation

The dominant theme in the subject of optimization is the use and allocation of processor registers. In the Oberon-0 compiler presented registers are used exclusively to hold anonymous intermediate results during the evaluation of expressions. For this purpose, usually a few registers suffice. Modern processors, however, feature a significant number of registers with access times considerably shorter than that of main memory. Using them for intermediate results only would imply a poor utilization of the most valuable resources. A primary goal of good code optimization is the most effective use of registers in order to reduce the number of accesses to the relatively slow main memory. A good strategy of register usage yields more advantages than any other branch of optimization.

A widespread technique is register allocation using graph colouring. For every value occurring in a computation, that is, for every expression the point of its generation and the point of its last use are determined. They delimit its range of relevance. Obviously, values of different expressions can be stored in the same register, if and only if their ranges do not overlap. The ranges are represented by the nodes of a graph, in which an edge between two nodes signifies that the two ranges overlap. The allocation of N available registers to the occurring values may then be understood as the colouring of the graph with N colours in such a way that neighbouring nodes always have different colours. This implies that values with overlapping ranges are always allocated to different registers.

Furthermore, selected, scalar, local variables are no longer allocated in memory at all, but rather in dedicated registers. In order to approach an optimal register utilization, sophisticated algorithms are employed to determine which variables are accessed most frequently. Evidently, the necessary

bookkeeping about variable accesses grows, and thereby compilation speed suffers. Also, care has to be taken that register values are saved in memory before procedure calls and are restored after the procedure return. The lurking danger is that the effort necessary for this task surpasses the savings obtained. In many compilers, local variables are allocated to registers only in procedures which do not contain any calls themselves (leaf procedures), and which therefore are also called most frequently, as they constitute the leaves in the tree representing the procedure call hierarchy.

A detailed treatment of all these optimization problems is beyond the scope of an introductory text about compiler construction. The above outline shall therefore suffice. In any case such techniques make it clear that for a nearly optimal code generation significantly more information about context must be considered than is the case in our relatively simple Oberon-0 compiler. Its structure is not well-suited to achieving a high degree of optimization. But it serves excellently as a fast compiler producing quite acceptable, although not optimal code, as is appropriate in the development phase of a system, particularly for educational purposes. Section 16.5 indicates another, somewhat more complex compiler structure which is better suited for the incorporation of optimization algorithms.

16.5. The frontend/backend compiler structure

The most significant characteristic of the compiler developed in Chapters 7 _ 12 is that the source text is read exactly once. Code is thereby generated on the fly. At each point, information about the operands is restricted to the items denoting the operand and to the symbol table representing declarations. The so-called frontend/backend compiler structure, which was briefly mentioned in Chapter 1, deviates decisively in this respect. The frontend part also reads the source text once only, but instead of generating code it builds a data structure representing the program in a form suitably organized for further processing. All information contained in statements is mapped into this data structure. It is called a syntax tree, because it also mirrors the syntactic structure of the text. Somewhat oversimplifying the situation, we may say that the frontend compiles declarations into the symbol table and statements into the syntax tree. These two data structures constitute the interface to the backend part whose task is code generation. The syntax tree allows fast access to practically all parts of a program, and it represents the program in a preprocessed form. The resulting compilation process is shown in Figure 16.1.




Figure 16.1. Compiler consisting of front end and back end

We pointed out one significant advantage of this structure in Chapter 1: the partitioning of a compiler in a target-independent front end and a target-dependent back end. In the following, we focus on the interface between the two parts, namely the structure of the syntax tree. Furthermore, we show how the tree is generated.

Exactly as in a source program where statements refer to declarations, so does the syntax tree refer to entries in the symbol table. This gives rise to the understandable desire to declare the elements of the symbol table (objects) in such a fashion that it is possible to refer to them from the symbol table itself as well as from the syntax tree. As basic type we introduce the type Object which may assume different forms as appropriate to represent constants, variables, types, and procedures. Only the attribute type is common to all. Here and subsequently we make use of Oberon's feature called type extension (Reiser and Wirth, 1992).

```

Object      = POINTER TO ObjDesc;
ObjDesc     = RECORD type: Type END ;
ConstDesc   = RECORD (ObjDesc) value: LONGINT END ;
VarDesc     = RECORD (ObjDesc) adr, level: LONGINT END ;

```

The symbol table consists of lists of elements, one for each scope (see Section 8.2). The elements consist of the name (identifier) and a reference to the identified object.

```

Ident       = POINTER TO IdentDesc;
IdentDesc   = RECORD name: ARRAY 32 OF CHAR;
              obj: Object; next: Ident
            END ;
Scope       = POINTER TO ScopeDesc;
ScopeDesc   = RECORD first: Ident; dsc: Scope END ;

```

The syntax tree is best conceived as a binary tree. We call its elements Nodes. If a syntactic construct has the form of a list, it is represented as a degenerate tree in which the last element has an empty branch.

```

Node =      POINTER TO NodeDesc;
NodeDesc = RECORD (Object)
              op: INTEGER;
              left, right: Object
            END

```

Let us consider the following brief excerpt of a program text as an example:

```

VAR x, y, z: INTEGER;
BEGIN z := x + y - 5; ...

```

The front end parses the source text and builds the symbol table and the syntax tree as shown in Figure 16.2. Representations of data types are omitted.




Figure 16.2. Symbol table (below) and syntax tree (above).

Representations of procedure calls, the IF and WHILE statements and the statement sequence are shown in Figures 16.3 - 16.5.




Figure 16.3. Procedure call.




Figure 16.4. IF and WHILE statements.




Figure 16.5. Statement sequence.

To conclude, the following examples demonstrate how the described data structures are generated. The reader should compare these compiler excerpts with the corresponding procedures of the Oberon-0 compiler listed in Appendix C. All subsequent algorithms make use of the auxiliary procedure `New`, which generates a new node.

```

PROCEDURE New(op: INTEGER; x, y: Object): Item;
  VAR z: Item;
BEGIN New(z); z.op := op; z.left := x; z.right := y; RETURN z
END New;

PROCEDURE factor(): Object;
  VAR x: Object; c: Constant;
BEGIN
  IF sym = ident THEN x := This(name); Get(sym); x := selector(x)
  ELSIF sym = number THEN NEW(c); c.value := number; Get(sym); x := c
  ELSIF sym = lparen THEN Get(sym); x := expression();
    IF sym = rparen THEN Get(sym) ELSE Mark(22) END
  ELSIF sym = not THEN Get(sym); x := New(not, NIL, factor())
  ELSE ...
  END ;
  RETURN x
END factor;

```

```

PROCEDURE term(): Object;
  VAR op: INTEGER; x: Object;
BEGIN x := factor();
  WHILE (sym >= times) & (sym <= and) DO
    op := sym; Get(sym); x := New(op, x, factor())
  END ;
  RETURN x
END term;

PROCEDURE statement(): Object;
  VAR x: Object;
BEGIN
  IF sym = ident THEN
    x := This(name); Get(sym); x := selector(x);
    IF sym = becomes THEN Get(sym); x := New(becomes, x, expression())
  ELSIF ...
  END
  ELSIF sym = while THEN
    Get(sym); x := expression();
    IF sym = do THEN Get(sym) ELSE Mark(25) END ;
    x := New(while, x, statseq());
    IF sym = end THEN Get(sym) ELSE Mark(20) END
  ELSIF ...
  END ;
  RETURN x
END statement

```

These excerpts clearly show that the structure of the front end is predetermined by the parser. The program has even become slightly simpler. But it must be kept in mind that type checking has been omitted in the above procedures for the sake of brevity. However, as a target-independent task, type checking clearly belongs to the front end.

16.6. Exercises

16.1. Improve code generation of the Oberon-0 compiler such that values and addresses, once loaded into a register, may possibly be reused without reloading. For the example

$$z := (x - y) * (x + y); y := x$$

the presented compiler generates the instruction sequence

```

LDW  1, 0, x
LDW  2, 0, y
SUB   1, 1, 2
LDW  2, 0, x
LDW  3, 0, y
ADD   2, 2, 3
MUL   1, 1, 2
STW   1, 0, z
LDW   1, 0, x
STW   1, 0, y

```

The improved version is to generate

```

LDW  1, 0, x
LDW  2, 0, y
SUB   3, 1, 2
ADD   4, 1, 2
MUL   5, 3, 4

```

```
STW    5, 0, z
STW    1, 0, y
```

Measure the gain on hand of a reasonably large number of test cases.

- 16.2. Which additional instructions of the RISC architecture of Chapter 9 would be desirable to facilitate the implementations of the preceding exercises, and to generate shorter and more efficient code?
- 16.3. Optimize the Oberon-0 compiler in such a way that scalar variables are allocated in registers instead of memory if possible. Measure the achieved gain and compare it with the one obtained in Exercise 16.1. How are variables treated as VAR parameters?
- 16.4. Construct a module OSGx which replaces OSG (see listing in Appendix C) and generates code for a CISC architecture x. The given interface of OSG should be retained as far as possible in order that modules OSS and OSP remain unchanged.

Appendix A

Syntax

A.1 Oberon-0

```
ident = letter {letter | digit}.
integer = digit {digit}.

selector = {"." ident | "[" expression ""]}.
factor = ident selector | integer | "(" expression ")" | "~" factor.
term = factor {"*" | "DIV" | "MOD" | "&") factor}.
SimpleExpression = ["+" | "-"] term {"+" | "- | "OR") term}.
expression = SimpleExpression
  [( "=" | "#" | "<" | "<=" | ">" | ">=") SimpleExpression].
assignment = ident selector ":=" expression.
ActualParameters = "(" [expression {"," expression}] ")" .
ProcedureCall = ident [ActualParameters].
IfStatement = "IF" expression "THEN" StatementSequence
  {"ELSIF" expression "THEN" StatementSequence}
  ["ELSE" StatementSequence] "END".
WhileStatement = "WHILE" expression "DO" StatementSequence "END".
statement = [assignment | ProcedureCall | IfStatement | WhileStatement].
StatementSequence = statement {";" statement}.
IdentList = ident {";" ident}.
ArrayType = "ARRAY" expression "OF" type.
FieldList = [IdentList ":" type].
RecordType = "RECORD" FieldList {";" FieldList} "END".
type = ident | ArrayType | RecordType.
FPSection = ["VAR"] IdentList ":" type.
FormalParameters = "(" [FPSection {";" FPSection}] ")".
ProcedureHeading = "PROCEDURE" ident [FormalParameters].
ProcedureBody = declarations ["BEGIN" StatementSequence] "END".
ProcedureDeclaration = ProcedureHeading ";" ProcedureBody ident.
declarations = ["CONST" {ident "=" expression ";"}]
  ["TYPE" {ident "=" type ";"}]
  ["VAR" {IdentList ":" type ";"}]
  {ProcedureDeclaration ";"} .
module = "MODULE" ident ";" declarations
  ["BEGIN" StatementSequence] "END" ident "." .
```

A.2 Oberon

```
ident = letter {letter | digit}.
number = integer | real.
integer = digit {digit} | digit {hexDigit} "H".
real = digit {digit} "." {digit} [ScaleFactor].
ScaleFactor = ("E" | "D") ["+" | "-"] digit {digit}.
hexDigit = digit | "A" | "B" | "C" | "D" | "E" | "F".
digit = "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9".
CharConstant = "" character "" | digit {hexDigit} "X".
string = "" {character} "".
identdef = ident ["*"].
qualident = [ident "."] ident.
ConstantDeclaration = identdef "=" ConstExpression.
ConstExpression = expression.
```



```

TypeDeclaration = identdef "=" type.
type = qualident | ArrayType | RecordType | PointerType | ProcedureType.
ArrayType = ARRAY length {" length} OF type.
length = ConstExpression.
RecordType = RECORD [" BaseType ") FieldListSequence END.
BaseType = qualident.
FieldListSequence = FieldList {" FieldList}.
FieldList = [IdentList ":" type].
IdentList = identdef {" identdef}.
PointerType = POINTER TO type.
ProcedureType = PROCEDURE [FormalParameters].
VariableDeclaration = IdentList ":" type.
    designator = qualident {" ident | "[" ExpList "]" | "(" qualident ")" | "^" }.
ExpList = expression {" expression}.
expression = SimpleExpression [relation SimpleExpression].
relation = "=" | "#" | "<" | "<=" | ">" | ">=" | IN | IS.
SimpleExpression = ["+" | "-"] term {AddOperator term}.
AddOperator = "+" | "-" | OR.
term = factor {MulOperator factor}.
MulOperator = "*" | "/" | DIV | MOD | "&".
factor = number | CharConstant | string | NIL | set |
    designator [ActualParameters] | "(" expression ")" | "~" factor.
set = "{" [element {" element}] "}".
element = expression [".. expression].
ActualParameters = "(" [ExpList] ")".
statement = [assignment | ProcedureCall |
    IfStatement | CaseStatement | WhileStatement | RepeatStatement |
    LoopStatement | ForStatement | WithStatement | EXIT | RETURN [expression] ].
assignment = designator ":" expression.
ProcedureCall = designator [ActualParameters].
StatementSequence = statement {" statement}.
IfStatement = IF expression THEN StatementSequence
    {ELSIF expression THEN StatementSequence}
    [ELSE StatementSequence] END.
CaseStatement = CASE expression OF case {" case}
    [ELSE StatementSequence] END.
case = [CaseLabelList ":" StatementSequence].
CaseLabelList = CaseLabels {" CaseLabels}.
CaseLabels = ConstExpression [".. ConstExpression].
WhileStatement = WHILE expression DO StatementSequence END.
RepeatStatement = REPEAT StatementSequence UNTIL expression.
LoopStatement = LOOP StatementSequence END.
ForStatement = FOR ident ":" expression TO expression [BY ConstExpression] DO
StatementSequence END.
WithStatement = WITH qualident ":" qualident DO StatementSequence END.

ProcedureDeclaration = ProcedureHeading ";" ProcedureBody ident.
ProcedureHeading = PROCEDURE ["*"] identdef [FormalParameters].
ProcedureBody = DeclarationSequence [BEGIN StatementSequence] END.
ForwardDeclaration = PROCEDURE "^" ident ["*"] [FormalParameters].
DeclarationSequence = {CONST {ConstantDeclaration ";" } |
    TYPE {TypeDeclaration ";" } | VAR {VariableDeclaration ";" } |
    {ProcedureDeclaration ";" } | ForwardDeclaration ";" }.
FormalParameters = "(" [FPSection {" FPSection}] ")" [" qualident].
FPSection = [VAR] ident {" ident} ":" FormalType.
FormalType = {ARRAY OF} (qualident | ProcedureType).

```

```

ImportList = IMPORT import {"," import} ",".
import = ident [":=" ident].
module = MODULE ident "," [ImportList] DeclarationSequence
        [BEGIN StatementSequence] END ident ".".

```

A.3 Symbol files

```

SymFile = BEGIN key {name key}   imported modules
        [ CONST {type name value} ] [ VAR {type name} ] constants and variables
        [ PROC {type name} {[VAR] type name} END} ]   procedures, parameters
        [ ALIAS {type name} ] [ NEWTYP {type} ] END . renamed procedures
type = basicType | [Module] OldType | NewType.
basicType = BOOL | CHAR | INTEGER | REAL | ...
NewType = ARRAY type name intval | DYNARRAY type name | POINTER type name
        | RECORD type name {type name} END   record types and fields
        | PROCTYP type name {[VAR] type name} END .   procedure types and parameters

```

Words consisting of upper-case letters denote terminal symbols. In the symbol file, they are encoded as integers. OldType and Module denote type and module numbers, that is, they are references to previously defined objects.

Appendix B

The ASCII Character Set

0	1	2	3	4	5	6	7	
0	nul	dle		0	@	P	`	p
1	soh	dc1	!	1	A	Q	a	q
2	stx	dc2	"	2	B	R	b	r
3	etx	dc3	#	3	C	S	c	s
4	eot	dc4	\$	4	D	T	d	t
5	enq	nak	%	5	E	U	e	u
6	ack	syn	&	6	F	V	f	v
7	bel	etb	'	7	G	W	g	w
8	bs	can	(8	H	X	h	x
9	ht	em)	9	I	Y	i	y
A	lf	sub	*	:	J	Z	j	z
B	vt	esc	+	;	K	[k	{
C	ff	fs	,	<	L	\	l	
D	cr	gs	-	=	M]	m	}
E	so	rs	.	>	N	^	n	~
F	si	us	/	?	O	_	o	del

Appendix C

The Oberon-0 Compiler

The compiler is split into three modules, namely the scanner OSS, the parser OSP, and the code generator OSG. Only OSG references the target architecture (see Chapter 9) by importing module RISC, which is an interpreter. With the aid of this interpreter, compiled code can be executed directly after compilation.

In Oberon, global, parameterless procedures are said to be commands, and they can be activated by the user via the Oberon operating system. The command Compile in the principal module OSP initiates the parsing process by a call of procedure Module which corresponds to the start symbol of the syntax. The text to be compiled is given as parameter. According to the conventions of the Oberon system, it can be specified in several ways:

OSP.Compile name	the source text is the file name
OSP.Compile *	the source text is the text in the marked viewer
OSP.Compile @	the source text starts with the latest selection

Successful compilation is immediately followed by the loading of the compiled code and execution of the compiled module's body. Here, loading is emulated by copying the instructions from array code in OSG into array M in RISC. The relative addresses of global variables are made absolute by adding a base value as is customary in program loaders. The compiler delivers the required data in the form of the table ref containing the addresses of all instructions to be fixed up.

Commands are registered by the compiler in tables comname and comadr. Procedure OSP.Exec name searches the name in the table comname, and it supplies the corresponding address from table comadr to the interpreter as the starting point for execution.

If a compiled procedure contains calls of Read, the numbers textually following the words OSP.Exec name are read and appear as inputs to the interpreter. For example, the procedure

```
PROCEDURE Add;
  VAR x, y, z: INTEGER;
  BEGIN Read(x); Read(y); Read(z); Write(x+y+z); WriteLn
  END Add
```

activated by the Oberon command *OSP.Exec Add 3 5 7* yields the output *15*.

C.1. The scanner

```
MODULE OSS; (* NW 19.9.93 / 17.11.94*)
  IMPORT Oberon, Texts;

  CONST IdLen* = 16; KW = 34;
  (*symbols*) null = 0;
  times* = 1; div* = 3; mod* = 4; and* = 5; plus* = 6; minus* = 7; or* = 8;
  eql* = 9; neq* = 10; lss* = 11; geq* = 12; leq* = 13; gtr* = 14;
  period* = 18; comma* = 19; colon* = 20; rparen* = 22; rbrak* = 23;
  of* = 25; then* = 26; do* = 27;
  lparen* = 29; lbrak* = 30; not* = 32; becomes* = 33; number* = 34; ident* = 37;
  semicolon* = 38; end* = 40; else* = 41; elsif* = 42;
  if* = 44; while* = 46;
  array* = 54; record* = 55;
  const* = 57; type* = 58; var* = 59; procedure* = 60; begin* = 61; module* = 63; eof = 64;

  TYPE Ident* = ARRAY IdLen OF CHAR;

  VAR val*: LONGINT;
  id*: Ident;
  error*: BOOLEAN;
  ch: CHAR;
  nkwl: INTEGER;
  errpos: LONGINT;
  R: Texts.Reader;
  W: Texts.Writer;
  keyTab : ARRAY KW OF
    RECORD sym: INTEGER; id: ARRAY 12 OF CHAR END;

  PROCEDURE Mark*(msg: ARRAY OF CHAR);
    VAR p: LONGINT;
  BEGIN p := Texts.Pos(R) - 1;
    IF p > errpos THEN
      Texts.WriteString(W, " pos "); Texts.WriteInt(W, p, 1);
      Texts.Write(W, " "); Texts.WriteString(W, msg);
```

```

    Texts.WriteLn(W); Texts.Append(Oberon.Log, W.buf)
END ;
errpos := p; error := TRUE
END Mark;

PROCEDURE Get*(VAR sym: INTEGER);

  PROCEDURE Ident;
    VAR i, k: INTEGER;
  BEGIN i := 0;
    REPEAT
      IF i < IdLen THEN id[i] := ch; INC(i) END ;
      Texts.Read(R, ch)
    UNTIL (ch < "0") OR (ch > "9") & (CAP(ch) < "A") OR (CAP(ch) > "Z");
    id[i] := 0X; k := 0;
    WHILE (k < nkW) & (id # keyTab[k].id) DO INC(k) END ;
    IF k < nkW THEN sym := keyTab[k].sym ELSE sym := ident END
  END Ident;

  PROCEDURE Number;
  BEGIN val := 0; sym := number;
    REPEAT
      IF val <= (MAX(LONGINT) - ORD(ch) + ORD("0")) DIV 10 THEN
        val := 10 * val + (ORD(ch) - ORD("0"))
      ELSE Mark("number too large"); val := 0
      END ;
      Texts.Read(R, ch)
    UNTIL (ch < "0") OR (ch > "9")
  END Number;

  PROCEDURE comment;
  BEGIN Texts.Read(R, ch);
    LOOP
      LOOP
        WHILE ch = "(" DO Texts.Read(R, ch);
          IF ch = "*" THEN comment END
        END ;
        IF ch = "*" THEN Texts.Read(R, ch); EXIT END ;
        IF R.eot THEN EXIT END ;
        Texts.Read(R, ch)
      END ;
      IF ch = ")" THEN Texts.Read(R, ch); EXIT END ;
      IF R.eot THEN Mark("comment not terminated"); EXIT END
    END
  END comment;

BEGIN
  WHILE ~R.eot & (ch <= " ") DO Texts.Read(R, ch) END;
  IF R.eot THEN sym := eof
  ELSE
    CASE ch OF
      "&": Texts.Read(R, ch); sym := and
    |   "*": Texts.Read(R, ch); sym := times
    |   "+": Texts.Read(R, ch); sym := plus
    |   "-": Texts.Read(R, ch); sym := minus
    |   "=": Texts.Read(R, ch); sym := eql
    |   "#": Texts.Read(R, ch); sym := neq
    |   "<": Texts.Read(R, ch);

```

```

        IF ch = "=" THEN Texts.Read(R, ch); sym := leq ELSE sym := lss END
    | ">": Texts.Read(R, ch);
        IF ch = "=" THEN Texts.Read(R, ch); sym := geq ELSE sym := gtr END
    | ";": Texts.Read(R, ch); sym := semicolon
    | ",": Texts.Read(R, ch); sym := comma
    | "(": Texts.Read(R, ch);
        IF ch = "=" THEN Texts.Read(R, ch); sym := becomes ELSE sym := colon END
    | ".": Texts.Read(R, ch); sym := period
    | "(": Texts.Read(R, ch);
        IF ch = "*" THEN comment; Get(sym) ELSE sym := lparen END
    | ")": Texts.Read(R, ch); sym := rparen
    | "[": Texts.Read(R, ch); sym := lbrak
    | "]": Texts.Read(R, ch); sym := rbrak
    | "0".."9": Number;
    | "A" .. "Z", "a" .. "z": Ident
    | "~": Texts.Read(R, ch); sym := not
    ELSE Texts.Read(R, ch); sym := null
    END
END
END Get;

PROCEDURE Init*(T: Texts.Text; pos: LONGINT);
BEGIN error := FALSE; errpos := pos; Texts.OpenReader(R, T, pos); Texts.Read(R, ch)
END Init;

PROCEDURE EnterKW(sym: INTEGER; name: ARRAY OF CHAR);
BEGIN keyTab[nkw].sym := sym; COPY(name, keyTab[nkw].id); INC(nkw)
END EnterKW;

BEGIN Texts.OpenWriter(W); error := TRUE; nkw := 0;
    EnterKW(null, "BY");
    EnterKW(do, "DO");
    EnterKW(if, "IF");
    EnterKW(null, "IN");
    EnterKW(null, "IS");
    EnterKW(of, "OF");
    EnterKW(or, "OR");
    EnterKW(null, "TO");
    EnterKW(end, "END");
    EnterKW(null, "FOR");
    EnterKW(mod, "MOD");
    EnterKW(null, "NIL");
    EnterKW(var, "VAR");
    EnterKW(null, "CASE");
    EnterKW(else, "ELSE");
    EnterKW(null, "EXIT");
    EnterKW(then, "THEN");
    EnterKW(type, "TYPE");
    EnterKW(null, "WITH");
    EnterKW(array, "ARRAY");
    EnterKW(begin, "BEGIN");
    EnterKW(const, "CONST");
    EnterKW(elsif, "ELSIF");
    EnterKW(null, "IMPORT");
    EnterKW(null, "UNTIL");
    EnterKW(while, "WHILE");
    EnterKW(record, "RECORD");

```

```

EnterKW(null, "REPEAT");
EnterKW(null, "RETURN");
EnterKW(null, "POINTER");
EnterKW(procedure, "PROCEDURE");
EnterKW(div, "DIV");
EnterKW(null, "LOOP");
EnterKW(module, "MODULE");
END OSS.

```

C.2. The parser

```

MODULE OSP; (* NW 23.9.93 / 9.2.95*)
  IMPORT Viewers, Texts, Oberon, MenuViewers, TextFrames, OSS, OSG;

  CONST WordSize = 4;
  VAR sym: INTEGER; loaded: BOOLEAN;
      topScope, universe: OSG.Object; (* linked lists, end with guard *)
      guard: OSG.Object;
      W: Texts.Writer;

  PROCEDURE NewObj(VAR obj: OSG.Object; class: INTEGER);
    VAR new, x: OSG.Object;
  BEGIN x := topScope; guard.name := OSS.id;
    WHILE x.next.name # OSS.id DO x := x.next END ;
    IF x.next = guard THEN
      NEW(new); new.name := OSS.id; new.class := class; new.next := guard;
      x.next := new; obj := new
    ELSE obj := x.next; OSS.Mark("mult def")
    END
  END NewObj;

  PROCEDURE find(VAR obj: OSG.Object);
    VAR s, x: OSG.Object;
  BEGIN s := topScope; guard.name := OSS.id;
    LOOP x := s.next;
      WHILE x.name # OSS.id DO x := x.next END ;
      IF x # guard THEN obj := x; EXIT END ;
      IF s = universe THEN obj := x; OSS.Mark("undef"); EXIT END ;
      s := s.dsc
    END
  END find;

  PROCEDURE FindField(VAR obj: OSG.Object; list: OSG.Object);
  BEGIN guard.name := OSS.id;
    WHILE list.name # OSS.id DO list := list.next END ;
    obj := list
  END FindField;

  PROCEDURE IsParam(obj: OSG.Object): BOOLEAN;
  BEGIN RETURN (obj.class = OSG.Par) OR (obj.class = OSG.Var) & (obj.val > 0)
  END IsParam;

  PROCEDURE OpenScope;
    VAR s: OSG.Object;
  BEGIN NEW(s); s.class := OSG.Head; s.dsc := topScope; s.next := guard; topScope := s
  END OpenScope;

```

```

PROCEDURE CloseScope;
BEGIN topScope := topScope.dsc
END CloseScope;

(* ----- Parser ----- *)

PROCEDURE^ expression(VAR x: OSG.Item);

PROCEDURE selector(VAR x: OSG.Item);
  VAR y: OSG.Item; obj: OSG.Object;
BEGIN
  WHILE (sym = OSS.lbrak) OR (sym = OSS.period) DO
    IF sym = OSS.lbrak THEN
      OSS.Get(sym); expression(y);
      IF x.type.form = OSG.Array THEN OSG.Index(x, y) ELSE OSS.Mark("not an array") END ;
      IF sym = OSS.rbrak THEN OSS.Get(sym) ELSE OSS.Mark("]") END
    ELSE OSS.Get(sym);
      IF sym = OSS.ident THEN
        IF x.type.form = OSG.Record THEN
          FindField(obj, x.type.fields); OSS.Get(sym);
          IF obj # guard THEN OSG.Field(x, obj) ELSE OSS.Mark("undef") END
        ELSE OSS.Mark("not a record")
        END
      ELSE OSS.Mark("ident?")
      END
    END
  END
END selector;

PROCEDURE factor(VAR x: OSG.Item);
  VAR obj: OSG.Object;
BEGIN (*sync*)
  IF sym < OSS.lparen THEN OSS.Mark("ident?");
    REPEAT OSS.Get(sym) UNTIL sym >= OSS.lparen
  END ;
  IF sym = OSS.ident THEN find(obj); OSS.Get(sym); OSG.MakeItem(x, obj); selector(x)
  ELSIF sym = OSS.number THEN OSG.MakeConstItem(x, OSG.intType, OSS.val); OSS.Get(sym)
  ELSIF sym = OSS.lparen THEN
    OSS.Get(sym); expression(x);
    IF sym = OSS.rparen THEN OSS.Get(sym) ELSE OSS.Mark(")?") END
  ELSIF sym = OSS.not THEN OSS.Get(sym); factor(x); OSG.Op1(OSS.not, x)
  ELSE OSS.Mark("factor?"); OSG.MakeItem(x, guard)
  END
END factor;

PROCEDURE term(VAR x: OSG.Item);
  VAR y: OSG.Item; op: INTEGER;
BEGIN factor(x);
  WHILE (sym >= OSS.times) & (sym <= OSS.and) DO
    op := sym; OSS.Get(sym);
    IF op = OSS.and THEN OSG.Op1(op, x) END ;
    factor(y); OSG.Op2(op, x, y)
  END
END term;

```

```

PROCEDURE SimpleExpression(VAR x: OSG.Item);
  VAR y: OSG.Item; op: INTEGER;
BEGIN
  IF sym = OSS.plus THEN OSS.Get(sym); term(x)
  ELSIF sym = OSS.minus THEN OSS.Get(sym); term(x); OSG.Op1(OSS.minus, x)
  ELSE term(x)
  END;
  WHILE (sym >= OSS.plus) & (sym <= OSS.or) DO
    op := sym; OSS.Get(sym);
    IF op = OSS.or THEN OSG.Op1(op, x) END ;
    term(y); OSG.Op2(op, x, y)
  END
END SimpleExpression;

```

```

PROCEDURE expression(VAR x: OSG.Item);
  VAR y: OSG.Item; op: INTEGER;
BEGIN SimpleExpression(x);
  IF (sym >= OSS.eql) & (sym <= OSS.gtr) THEN
    op := sym; OSS.Get(sym); SimpleExpression(y); OSG.Relation(op, x, y)
  END
END expression;

```

```

PROCEDURE parameter(VAR fp: OSG.Object);
  VAR x: OSG.Item;
BEGIN expression(x);
  IF IsParam(fp) THEN OSG.Parameter(x, fp.type, fp.class); fp := fp.next
  ELSE OSS.Mark("too many parameters")
  END
END parameter;

```

```

PROCEDURE StatSequence;
  VAR par, obj: OSG.Object; x, y: OSG.Item; L: LONGINT;

```

```

  PROCEDURE param(VAR x: OSG.Item);
  BEGIN
    IF sym = OSS.lparen THEN OSS.Get(sym) ELSE OSS.Mark(")?") END ;
    expression(x);
    IF sym = OSS.rparen THEN OSS.Get(sym) ELSE OSS.Mark(")?") END
  END param;

```

```

BEGIN (* StatSequence *)
  LOOP (*sync*) obj := guard;
    IF sym < OSS.ident THEN OSS.Mark("statement?");
      REPEAT OSS.Get(sym) UNTIL sym >= OSS.ident
    END ;
    IF sym = OSS.ident THEN
      find(obj); OSS.Get(sym); OSG.MakeItem(x, obj); selector(x);
      IF sym = OSS.becomes THEN OSS.Get(sym); expression(y); OSG.Store(x, y)
      ELSIF sym = OSS.eql THEN OSS.Mark(":= ?"); OSS.Get(sym); expression(y)
      ELSIF x.mode = OSG.Proc THEN
        par := obj.dsc;
        IF sym = OSS.lparen THEN OSS.Get(sym);
          IF sym = OSS.rparen THEN OSS.Get(sym)
        ELSE
          LOOP parameter(par);
            IF sym = OSS.comma THEN OSS.Get(sym)

```



```

        ELSIF sym = OSS.rparen THEN OSS.Get(sym); EXIT
        ELSIF sym >= OSS.semicolon THEN EXIT
        ELSE OSS.Mark(" or , ?")
        END
    END
END
END ;
IF obj.val < 0 THEN OSS.Mark("forward call")
ELSIF ~IsParam(par) THEN OSG.Call(x)
ELSE OSS.Mark("too few parameters")
END
ELSIF x.mode = OSG.SProc THEN
    IF obj.val <= 3 THEN param(y) END ;
    OSG.IOCall(x, y)
ELSIF obj.class = OSG.Typ THEN OSS.Mark("illegal assignment?")
ELSE OSS.Mark("statement?")
END
ELSIF sym = OSS.if THEN
    OSS.Get(sym); expression(x); OSG.CJump(x);
    IF sym = OSS.then THEN OSS.Get(sym) ELSE OSS.Mark("THEN?") END ;
    StatSequence; L := 0;
    WHILE sym = OSS.elsif DO
        OSS.Get(sym); OSG.FJump(L); OSG.FixLink(x.a); expression(x); OSG.CJump(x);
        IF sym = OSS.then THEN OSS.Get(sym) ELSE OSS.Mark("THEN?") END ;
        StatSequence
    END ;
    IF sym = OSS.else THEN
        OSS.Get(sym); OSG.FJump(L); OSG.FixLink(x.a); StatSequence
    ELSE OSG.FixLink(x.a)
    END ;
    OSG.FixLink(L);
    IF sym = OSS.end THEN OSS.Get(sym) ELSE OSS.Mark("END?") END
ELSIF sym = OSS.while THEN
    OSS.Get(sym); L := OSG.pc; expression(x); OSG.CJump(x);
    IF sym = OSS.do THEN OSS.Get(sym) ELSE OSS.Mark("DO?") END ;
    StatSequence; OSG.BJump(L); OSG.FixLink(x.a);
    IF sym = OSS.end THEN OSS.Get(sym) ELSE OSS.Mark("END?") END
END ;
IF sym = OSS.semicolon THEN OSS.Get(sym)
ELSIF (sym >= OSS.semicolon) & (sym < OSS.if) OR (sym >= OSS.array) THEN EXIT
ELSE OSS.Mark(" ; ?")
END
END
END StatSequence;

PROCEDURE IdentList(class: INTEGER; VAR first: OSG.Object);
    VAR obj: OSG.Object;
BEGIN
    IF sym = OSS.ident THEN
        NewObj(first, class); OSS.Get(sym);
        WHILE sym = OSS.comma DO
            OSS.Get(sym);
            IF sym = OSS.ident THEN NewObj(obj, class); OSS.Get(sym)
            ELSE OSS.Mark("ident?")
            END
        END
    END;
END;

```

```

    IF sym = OSS.colon THEN OSS.Get(sym) ELSE OSS.Mark("?:") END
END
END IdentList;

PROCEDURE Type(VAR type: OSG.Type);
    VAR obj, first: OSG.Object; x: OSG.Item; tp: OSG.Type;
BEGIN type := OSG.intType; (*sync*)
    IF (sym # OSS.ident) & (sym < OSS.array) THEN OSS.Mark("type?");
        REPEAT OSS.Get(sym) UNTIL (sym = OSS.ident) OR (sym >= OSS.array)
    END ;
    IF sym = OSS.ident THEN
        find(obj); OSS.Get(sym);
        IF obj.class = OSG.Type THEN type := obj.type ELSE OSS.Mark("type?") END
    ELSIF sym = OSS.array THEN
        OSS.Get(sym); expression(x);
        IF (x.mode # OSG.Const) OR (x.a < 0) THEN OSS.Mark("bad index") END ;
        IF sym = OSS.of THEN OSS.Get(sym) ELSE OSS.Mark("OF?") END ;
        Type(tp); NEW(type); type.form := OSG.Array; type.base := tp;
        type.len := SHORT(x.a); type.size := type.len * tp.size
    ELSIF sym = OSS.record THEN
        OSS.Get(sym); NEW(type); type.form := OSG.Record; type.size := 0; OpenScope;
        LOOP
            IF sym = OSS.ident THEN
                IdentList(OSG.Fld, first); Type(tp); obj := first;
                WHILE obj # guard DO
                    obj.type := tp; obj.val := type.size; INC(type.size, obj.type.size); obj := obj.next
                END
            END ;
            IF sym = OSS.semicolon THEN OSS.Get(sym)
            ELSIF sym = OSS.ident THEN OSS.Mark("; ?")
            ELSE EXIT
            END
        END ;
        type.fields := topScope.next; CloseScope;
        IF sym = OSS.end THEN OSS.Get(sym) ELSE OSS.Mark("END?") END
    ELSE OSS.Mark("ident?")
    END
END Type;

PROCEDURE declarations(VAR varsize: LONGINT);
    VAR obj, first: OSG.Object;
        x: OSG.Item; tp: OSG.Type; L: LONGINT;
BEGIN (*sync*)
    IF (sym < OSS.const) & (sym # OSS.end) THEN OSS.Mark("declaration?");
        REPEAT OSS.Get(sym) UNTIL (sym >= OSS.const) OR (sym = OSS.end)
    END ;
    LOOP
        IF sym = OSS.const THEN
            OSS.Get(sym);
            WHILE sym = OSS.ident DO
                NewObj(obj, OSG.Const); OSS.Get(sym);
                IF sym = OSS.eql THEN OSS.Get(sym) ELSE OSS.Mark("=?") END;
                expression(x);
                IF x.mode = OSG.Const THEN obj.val := x.a; obj.type := x.type
                ELSE OSS.Mark("expression not constant")
                END;
            END
        END
    END
END declarations;

```

```

        IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark(";?") END
    END
END ;
IF sym = OSS.type THEN
    OSS.Get(sym);
    WHILE sym = OSS.ident DO
        NewObj(obj, OSG.Type); OSS.Get(sym);
        IF sym = OSS.eql THEN OSS.Get(sym) ELSE OSS.Mark("=?") END ;
        Type(obj.type);
        IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark(";?") END
    END
END ;
IF sym = OSS.var THEN
    OSS.Get(sym);
    WHILE sym = OSS.ident DO
        IdentList(OSG.Var, first); Type(tp); obj := first;
        WHILE obj # guard DO
            obj.type := tp; obj.lev := OSG.curlev;
            varsize := varsize + obj.type.size; obj.val := -varsize; obj := obj.next
        END ;
        IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark("; ?") END
    END
END ;
IF (sym >= OSS.const) & (sym <= OSS.var) THEN OSS.Mark("declaration?") ELSE EXIT END
END
END declarations;

PROCEDURE ProcedureDecl;
    CONST marksize = 8;
    VAR proc, obj: OSG.Object;
    procid: OSS.Ident;
    locblksize, parblksize: LONGINT;

PROCEDURE FPSection;
    VAR obj, first: OSG.Object; tp: OSG.Type; parsize: LONGINT;
BEGIN
    IF sym = OSS.var THEN OSS.Get(sym); IdentList(OSG.Par, first)
    ELSE IdentList(OSG.Var, first)
    END ;
    IF sym = OSS.ident THEN
        find(obj); OSS.Get(sym);
        IF obj.class = OSG.Type THEN tp := obj.type
        ELSE OSS.Mark("ident?"); tp := OSG.intType
        END
    ELSE OSS.Mark("ident?"); tp := OSG.intType
    END ;
    IF first.class = OSG.Var THEN
        parsize := tp.size;
        IF tp.form >= OSG.Array THEN OSS.Mark("no struct params") END ;
    ELSE parsize := WordSize
    END ;
    obj := first;
    WHILE obj # guard DO obj.type := tp; INC(parblksize, parsize); obj := obj.next END
END FPSection;

BEGIN (* ProcedureDecl *)

```

```

OSS.Get(sym);
IF sym = OSS.ident THEN
  procid := OSS.id;
  NewObj(proc, OSG.Proc); OSS.Get(sym); parblksize := marksize;
  OSG.IncLevel(1); OpenScope; proc.val := -1;
  IF sym = OSS.lparen THEN
    OSS.Get(sym);
    IF sym = OSS.rparen THEN OSS.Get(sym)
    ELSE FPSection;
      WHILE sym = OSS.semicolon DO OSS.Get(sym); FPSection END ;
      IF sym = OSS.rparen THEN OSS.Get(sym) ELSE OSS.Mark("?") END
    END
  ELSIF OSG.curlev = 1 THEN OSG.EnterCmd(procid)
  END ;
  obj := topScope.next; locblksize := parblksize;
  WHILE obj # guard DO
    obj.lev := OSG.curlev;
    IF obj.class = OSG.Par THEN DEC(locblksize, WordSize)
    ELSE obj.val := locblksize; obj := obj.next
    END
  END ;
  proc.dsc := topScope.next;
  IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark(";") END;
  locblksize := 0; declarations(locblksize);
  WHILE sym = OSS.procedure DO
    ProcedureDecl;
    IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark(";") END
  END ;
  proc.val := OSG.pc; OSG.Enter(locblksize);
  IF sym = OSS.begin THEN OSS.Get(sym); StatSequence END ;
  IF sym = OSS.end THEN OSS.Get(sym) ELSE OSS.Mark("END?") END ;
  IF sym = OSS.ident THEN
    IF procid # OSS.id THEN OSS.Mark("no match") END ;
    OSS.Get(sym)
  END ;
  OSG.Return(parblksize - marksize); CloseScope; OSG.IncLevel(-1)
END
END ProcedureDecl;

PROCEDURE Module(VAR S: Texts.Scanner);
  VAR modid: OSS.Ident; varsize: LONGINT;
BEGIN Texts.WriteString(W, " compiling ");
  IF sym = OSS.module THEN
    OSS.Get(sym); OSG.Open; OpenScope; varsize := 0;
    IF sym = OSS.ident THEN
      modid := OSS.id; OSS.Get(sym);
      Texts.WriteString(W, modid); Texts.WriteLine(W); Texts.Append(Oberon.Log, W.buf)
    ELSE OSS.Mark("ident?")
    END ;
    IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark(";") END;
    declarations(varsize);
    WHILE sym = OSS.procedure DO
      ProcedureDecl;
      IF sym = OSS.semicolon THEN OSS.Get(sym) ELSE OSS.Mark(";") END
    END ;
    OSG.Header(varsize);

```

```

IF sym = OSS.begin THEN OSS.Get(sym); StatSequence END ;
IF sym = OSS.end THEN OSS.Get(sym) ELSE OSS.Mark("END?") END ;
IF sym = OSS.ident THEN
  IF modid # OSS.id THEN OSS.Mark("no match") END ;
  OSS.Get(sym)
ELSE OSS.Mark("ident?")
END ;
IF sym # OSS.period THEN OSS.Mark(". ?") END ;
CloseScope;
IF ~OSS.error THEN
  COPY(modid, S.s); OSG.Close(S, varsize); Texts.WriteString(W, "code generated");
  Texts.WriteInt(W, OSG.pc, 6); Texts.WriteLine(W); Texts.Append(Oberon.Log, W.buf)
END
ELSE OSS.Mark("MODULE?")
END
END Module;

PROCEDURE Compile*;
  VAR beg, end, time: LONGINT;
  S: Texts.Scanner; T: Texts.Text; v: Viewers.Viewer;
BEGIN loaded := FALSE;
  Texts.OpenScanner(S, Oberon.Par.text, Oberon.Par.pos); Texts.Scan(S);
  IF S.class = Texts.Char THEN
    IF S.c = "*" THEN
      v := Oberon.MarkedViewer();
      IF (v.dsc # NIL) & (v.dsc.next IS TextFrames.Frame) THEN
        OSS.Init(v.dsc.next(TextFrames.Frame).text, 0); OSS.Get(sym); Module(S) END
      ELSIF S.c = "@" THEN
        Oberon.GetSelection(T, beg, end, time);
        IF time >= 0 THEN OSS.Init(T, beg); OSS.Get(sym); Module(S) END
      END
    ELSIF S.class = Texts.Name THEN
      NEW(T); Texts.Open(T, S.s); OSS.Init(T, 0); OSS.Get(sym); Module(S)
    END
  END
END Compile;

PROCEDURE Decode*;
  VAR V: MenuViewers.Viewer; T: Texts.Text;
  X, Y: INTEGER;
BEGIN T := TextFrames.Text(""); Oberon.AllocateSystemViewer(Oberon.Par.frame.X, X, Y);
  V := MenuViewers.New(
    TextFrames.NewMenu("Log.Text", "System.Close System.Copy System.Grow Edit.Search
Edit.Store"),
    TextFrames.NewText(T, 0), TextFrames.menuH, X, Y);
  OSG.Decode(T)
END Decode;

PROCEDURE Load*;
  VAR S: Texts.Scanner;
BEGIN
  IF ~OSS.error & ~loaded THEN
    Texts.OpenScanner(S, Oberon.Par.text, Oberon.Par.pos); OSG.Load(S); loaded := TRUE
  END
END Load;

PROCEDURE Exec*;

```

```

    VAR S: Texts.Scanner;
BEGIN
    IF loaded THEN
        Texts.OpenScanner(S, Oberon.Par.text, Oberon.Par.pos); Texts.Scan(S);
        IF S.class = Texts.Name THEN OSG.Exec(S) END
    END
END Exec;

PROCEDURE enter(cl: INTEGER; n: LONGINT; name: OSS.Ident; type: OSG.Type);
    VAR obj: OSG.Object;
BEGIN NEW(obj);
    obj.class := cl; obj.val := n; obj.name := name; obj.type := type; obj.dsc := NIL;
    obj.next := topScope.next; topScope.next := obj
END enter;

BEGIN Texts.OpenWriter(W); Texts.WriteString(W, "Oberon0 Compiler 9.2.95");
    Texts.WriteLine(W); Texts.Append(Oberon.Log, W.buf);
    NEW(guard); guard.class := OSG.Var; guard.type := OSG.intType; guard.val := 0;
    topScope := NIL; OpenScope;
    enter(OSG.Typ, 1, "BOOLEAN", OSG.boolType);
    enter(OSG.Typ, 2, "INTEGER", OSG.intType);
    enter(OSG.Const, 1, "TRUE", OSG.boolType);
    enter(OSG.Const, 0, "FALSE", OSG.boolType);
    enter(OSG.SProc, 1, "Read", NIL);
    enter(OSG.SProc, 2, "Write", NIL);
    enter(OSG.SProc, 3, "WriteHex", NIL);
    enter(OSG.SProc, 4, "WriteLn", NIL);
    universe := topScope
END OSP.

```

C.3. The code generator

```

MODULE OSG; (* NW 18.12.94 / 10.2.95 / 24.3.96 / 25.11.05*)
    IMPORT Oberon, Texts, OSS, RISC;

    CONST maxCode = 1000; maxRel = 200; NofCom = 16;
    (* class / mode*) Head* = 0;
        Var* = 1; Par* = 2; Const* = 3; Fld* = 4; Typ* = 5; Proc* = 6; SProc* = 7;
        Reg = 10; Cond = 11;
    (* form *) Boolean* = 0; Integer* = 1; Array* = 2; Record* = 3;

    MOV = 0; MVN = 1; ADD = 2; SUB = 3; MUL = 4; Div = 5; Mod = 6; CMP = 7;
    MOVI = 16; MVNI = 17; ADDI = 18; SUBI = 19; MULI = 20; DIVI = 21; MODI = 22; CMPI = 23;
    CHKI = 24;
    LDW = 32; LDB = 33; POP = 34; STW = 36; STB = 37; PSH = 38;
    RD = 40; WRD = 41; WRH = 42; WRL = 43;
    BEQ = 48; BNE = 49; BLT = 50; BGE = 51; BLE = 52; BGT = 53; BR = 56; BSR = 57; RET = 58;

    FP = 12; SP = 13; LNK = 14; PC = 15; (*reserved registers*)

    TYPE Object* = POINTER TO ObjDesc;
        Type* = POINTER TO TypeDesc;

    Item* = RECORD
        mode*, lev*: INTEGER;

```

```

    type*: Type;
    a*, b, c, r: LONGINT;
END ;

ObjDesc*= RECORD
    class*, lev*: INTEGER;
    next*, dsc*: Object;
    type*: Type;
    name*: OSS.Ident;
    val*: LONGINT
END ;

TypeDesc* = RECORD
    form*: INTEGER;
    fields*: Object;
    base*: Type;
    size*, len*: INTEGER
END ;

VAR boolType*, intType*: Type;
    curlev*, pc*: INTEGER;
    cno: INTEGER;
    entry, fixlist: LONGINT;
    regs: SET; (* used registers *)
    W: Texts.Writer;
    code: ARRAY maxCode OF LONGINT;
    comname: ARRAY NofCom OF OSS.Ident; (*commands*)
    comadr: ARRAY NofCom OF LONGINT;
    mnemo: ARRAY 64, 5 OF CHAR; (*for decoder*)

PROCEDURE GetReg(VAR r: LONGINT);
    VAR i: INTEGER;
BEGIN i := 0;
    WHILE (i < FP) & (i IN regs) DO INC(i) END ;
    INCL(regs, i); r := i
END GetReg;

PROCEDURE Put(op, a, b, c: LONGINT);
BEGIN (*emit instruction*)
    IF op >= 32 THEN DEC(op, 64) END ;
    code[pc] := ASH(ASH(ASH(op, 4) + a, 4) + b, 18) + (c MOD 40000H);
    INC(pc)
END Put;

PROCEDURE PutBR(op, disp: LONGINT);
BEGIN (*emit branch instruction*)
    code[pc] := ASH(op-40H, 26) + (disp MOD 4000000H); INC(pc)
END PutBR;

PROCEDURE TestRange(x: LONGINT);
BEGIN (*18-bit entity*)
    IF (x >= 20000H) OR (x < -20000H) THEN OSS.Mark("value too large") END
END TestRange;

PROCEDURE load(VAR x: Item);
    VAR r: LONGINT;

```

```

BEGIN (*x.mode # Reg*)
  IF x.mode = Var THEN
    IF x.lev = 0 THEN x.a := x.a - pc*4 END ;
    GetReg(r); Put(LDW, r, x.r, x.a); EXCL(regs, x.r); x.r := r
  ELSIF x.mode = Const THEN
    TestRange(x.a); GetReg(x.r); Put(MOVI, x.r, 0, x.a)
  END ;
  x.mode := Reg
END load;

PROCEDURE loadBool(VAR x: Item);
BEGIN
  IF x.type.form # Boolean THEN OSS.Mark("Boolean?") END ;
  load(x); x.mode := Cond; x.a := 0; x.b := 0; x.c := 1
END loadBool;

PROCEDURE PutOp(cd: LONGINT; VAR x, y: Item);
BEGIN
  IF x.mode # Reg THEN load(x) END ;
  IF y.mode = Const THEN TestRange(y.a); Put(cd+16, x.r, x.r, y.a)
  ELSE
    IF y.mode # Reg THEN load(y) END ;
    Put(cd, x.r, x.r, y.r); EXCL(regs, y.r)
  END
END PutOp;

PROCEDURE negated(cond: LONGINT): LONGINT;
BEGIN
  IF ODD(cond) THEN RETURN cond-1 ELSE RETURN cond+1 END
END negated;

PROCEDURE merged(L0, L1: LONGINT): LONGINT;
  VAR L2, L3: LONGINT;
BEGIN
  IF L0 # 0 THEN
    L2 := L0;
    LOOP L3 := code[L2] MOD 40000H;
      IF L3 = 0 THEN EXIT END ;
      L2 := L3
    END ;
    code[L2] := code[L2] - L3 + L1; RETURN L0
  ELSE RETURN L1
  END
END merged;

PROCEDURE fix(at, with: LONGINT);
BEGIN code[at] := code[at] DIV 400000H * 400000H + (with MOD 400000H)
END fix;

PROCEDURE FixWith(L0, L1: LONGINT);
  VAR L2: LONGINT;
BEGIN
  WHILE L0 # 0 DO L2 := code[L0] MOD 40000H; fix(L0, L1-L0); L0 := L2 END
END FixWith;

PROCEDURE FixLink*(L: LONGINT);

```



```

    VAR L1: LONGINT;
BEGIN
    WHILE L # 0 DO L1 := code[L] MOD 40000H; fix(L, pc-L); L := L1 END
END FixLink;

(*-----*)

PROCEDURE IncLevel*(n: INTEGER);
BEGIN INC(curlev, n)
END IncLevel;

PROCEDURE MakeConstItem*(VAR x: Item; typ: Type; val: LONGINT);
BEGIN x.mode := Const; x.type := typ; x.a := val
END MakeConstItem;

PROCEDURE MakeItem*(VAR x: Item; y: Object);
    VAR r: LONGINT;
BEGIN x.mode := y.class; x.type := y.type; x.lev := y.lev; x.a := y.val; x.b := 0;
    IF y.lev = 0 THEN x.r := PC
    ELSIF y.lev = curlev THEN x.r := FP
    ELSE OSS.Mark("level!"); x.r := 0
    END ;
    IF y.class = Par THEN GetReg(r); Put(LDW, r, x.r, x.a); x.mode := Var; x.r := r; x.a := 0 END
END MakeItem;

PROCEDURE Field*(VAR x: Item; y: Object); (* x := x.y *)
BEGIN INC(x.a, y.val); x.type := y.type
END Field;

PROCEDURE Index*(VAR x, y: Item); (* x := x[y] *)
BEGIN
    IF y.type # intType THEN OSS.Mark("index not integer") END ;
    IF y.mode = Const THEN
        IF (y.a < 0) OR (y.a >= x.type.len) THEN OSS.Mark("bad index") END ;
        INC(x.a, y.a * x.type.base.size)
    ELSE
        IF y.mode # Reg THEN load(y) END ;
        Put(CHKI, y.r, 0, x.type.len);
        Put(MULI, y.r, y.r, x.type.base.size);
        Put(ADD, y.r, x.r, y.r); EXCL(regs, x.r); x.r := y.r
    END;
    x.type := x.type.base
END Index;

PROCEDURE Op1*(op: INTEGER; VAR x: Item); (* x := op x *)
    VAR t: LONGINT;
BEGIN
    IF op = OSS.minus THEN
        IF x.type.form # Integer THEN OSS.Mark("bad type")
        ELSIF x.mode = Const THEN x.a := -x.a
        ELSE
            IF x.mode = Var THEN load(x) END ;
            Put(MVN, x.r, 0, x.r)
        END
    ELSIF op = OSS.not THEN
        IF x.mode # Cond THEN loadBool(x) END ;

```

```

    x.c := negated(x.c); t := x.a; x.a := x.b; x.b := t
ELSIF op = OSS.and THEN
    IF x.mode # Cond THEN loadBool(x) END ;
    PutBR(BEQ + negated(x.c), x.a); EXCL(regs, x.r); x.a := pc-1; FixLink(x.b); x.b := 0
ELSIF op = OSS.or THEN
    IF x.mode # Cond THEN loadBool(x) END ;
    PutBR(BEQ + x.c, x.b); EXCL(regs, x.r); x.b := pc-1; FixLink(x.a); x.a := 0
END
END Op1;

```

```

PROCEDURE Op2*(op: INTEGER; VAR x, y: Item); (* x := x op y *)
BEGIN
    IF (x.type.form = Integer) & (y.type.form = Integer) THEN
        IF (x.mode = Const) & (y.mode = Const) THEN
            (*overflow checks missing*)
            IF op = OSS.plus THEN INC(x.a, y.a)
            ELSIF op = OSS.minus THEN DEC(x.a, y.a)
            ELSIF op = OSS.times THEN x.a := x.a * y.a
            ELSIF op = OSS.div THEN x.a := x.a DIV y.a
            ELSIF op = OSS.mod THEN x.a := x.a MOD y.a
            ELSE OSS.Mark("bad type")
            END
        ELSE
            IF op = OSS.plus THEN PutOp(ADD, x, y)
            ELSIF op = OSS.minus THEN PutOp(SUB, x, y)
            ELSIF op = OSS.times THEN PutOp(MUL, x, y)
            ELSIF op = OSS.div THEN PutOp(Div, x, y)
            ELSIF op = OSS.mod THEN PutOp(Mod, x, y)
            ELSE OSS.Mark("bad type")
            END
        END
    ELSIF (x.type.form = Boolean) & (y.type.form = Boolean) THEN
        IF y.mode # Cond THEN loadBool(y) END ;
        IF op = OSS.or THEN x.a := y.a; x.b := merged(y.b, x.b); x.c := y.c
        ELSIF op = OSS.and THEN x.a := merged(y.a, x.a); x.b := y.b; x.c := y.c
        END
    ELSE OSS.Mark("bad type")
    END ;
END Op2;

```

```

PROCEDURE Relation*(op: INTEGER; VAR x, y: Item); (* x := x ? y *)
BEGIN
    IF (x.type.form # Integer) OR (y.type.form # Integer) THEN OSS.Mark("bad type")
    ELSE PutOp(CMP, x, y); x.c := op - OSS.eql; EXCL(regs, y.r)
    END ;
    x.mode := Cond; x.type := boolType; x.a := 0; x.b := 0
END Relation;

```

```

PROCEDURE Store*(VAR x, y: Item); (* x := y *)
    VAR r: LONGINT;
BEGIN
    IF (x.type.form IN {Boolean, Integer}) & (x.type.form = y.type.form) THEN
        IF y.mode = Cond THEN
            Put(BEQ + negated(y.c), y.r, 0, y.a); EXCL(regs, y.r); y.a := pc-1;
            FixLink(y.b); GetReg(y.r); Put(MOVI, y.r, 0, 1); PutBR(BR, 2);
            FixLink(y.a); Put(MOVI, y.r, 0, 0)

```

```

    ELSIF y.mode # Reg THEN load(y)
    END ;
    IF x.mode = Var THEN
        IF x.lev = 0 THEN x.a := x.a - pc*4 END ;
        Put(STW, y.r, x.r, x.a)
    ELSE OSS.Mark("illegal assignment")
    END ;
    EXCL(regs, x.r); EXCL(regs, y.r)
    ELSE OSS.Mark("incompatible assignment")
    END
END Store;

```

```

PROCEDURE Parameter*(VAR x: Item; ftyp: Type; class: INTEGER);
    VAR r: LONGINT;
BEGIN
    IF x.type = ftyp THEN
        IF class = Par THEN (*Var param*)
            IF x.mode = Var THEN
                IF x.a # 0 THEN GetReg(r); Put(ADDI, r, x.r, x.a)
                ELSE r := x.r
                END
            ELSE OSS.Mark("illegal parameter mode")
            END ;
            Put(PSH, r, SP, 4); EXCL(regs, r)
        ELSE (*value param*)
            IF x.mode # Reg THEN load(x) END ;
            Put(PSH, x.r, SP, 4); EXCL(regs, x.r)
            END
        ELSE OSS.Mark("bad parameter type")
        END
    END Parameter;

```

(*-----*)

```

PROCEDURE CJump*(VAR x: Item);
BEGIN
    IF x.type.form = Boolean THEN
        IF x.mode # Cond THEN loadBool(x) END ;
        PutBR(BEQ + negated(x.c), x.a); FixLink(x.b); x.a := pc-1
    ELSE OSS.Mark("Boolean?"); x.a := pc
    END
END CJump;

```

```

PROCEDURE BJump*(L: LONGINT);
BEGIN PutBR(BR, L-pc)
END BJump;

```

```

PROCEDURE FJump*(VAR L: LONGINT);
BEGIN PutBR(BR, L); L := pc-1
END FJump;

```

```

PROCEDURE Call*(VAR x: Item);
BEGIN PutBR(BSR, x.a - pc)
END Call;

```

```

PROCEDURE IOCall*(VAR x, y: Item);

```

```

    VAR z: Item;
BEGIN
    IF x.a < 4 THEN
        IF y.type.form # Integer THEN OSS.Mark("Integer?") END
    END ;
    IF x.a = 1 THEN
        GetReg(z.r); z.mode := Reg; z.type := intType; Put(RD, z.r, 0, 0); Store(y, z)
    ELSIF x.a = 2 THEN load(y); Put(WRD, 0, 0, y.r); EXCL(regs, y.r)
    ELSIF x.a = 3 THEN load(y); Put(WRH, 0, 0, y.r); EXCL(regs, y.r)
    ELSE Put(WRL, 0, 0, 0)
    END
END IOCall;

PROCEDURE Header*(size: LONGINT);
BEGIN entry := pc; Put(MOVI, SP, 0, RISC.MemSize - size); (*init SP*)
    Put(PSH, LNK, SP, 4)
END Header;

PROCEDURE Enter*(size: LONGINT);
BEGIN
    Put(PSH, LNK, SP, 4);
    Put(PSH, FP, SP, 4);
    Put(MOV, FP, 0, SP);
    Put(SUBL, SP, SP, size)
END Enter;

PROCEDURE Return*(size: LONGINT);
BEGIN
    Put(MOV, SP, 0, FP);
    Put(POP, FP, SP, 4);
    Put(POP, LNK, SP, size+4);
    PutBR(RET, LNK)
END Return;

PROCEDURE Open*;
BEGIN curlev := 0; pc := 0; cno := 0; regs := {}
END Open;

PROCEDURE Close*(VAR S: Texts.Scanner; globals: LONGINT);
BEGIN Put(POP, LNK, SP, 4); PutBR(RET, LNK);
END Close;

PROCEDURE EnterCmd*(VAR name: ARRAY OF CHAR);
BEGIN COPY(name, comname[cno]); comadr[cno] := pc*4; INC(cno)
END EnterCmd;

(*-----*)

PROCEDURE Load*(VAR S: Texts.Scanner);
BEGIN RISC.Load(code, pc);
    Texts.WriteString(W, " code loaded"); Texts.WriteLine(W); Texts.Append(Oberon.Log, W.buf);
    RISC.Execute(entry*4, S, Oberon.Log)
END Load;

PROCEDURE Exec*(VAR S: Texts.Scanner);
    VAR i: INTEGER;

```

```

BEGIN i := 0;
  WHILE (i < cno) & (S.s # comname[i]) DO INC(i) END ;
  IF i < cno THEN RISC.Execute(comadr[i], S, Oberon.Log) END
END Exec;

PROCEDURE Decode*(T: Texts.Text);
  VAR i, w, op, a: LONGINT;
BEGIN Texts.WriteString(W, "entry"); Texts.WriteInt(W, entry*4, 6); Texts.WriteLn(W);
  i := 0;
  WHILE i < pc DO
    w := code[i]; op := w DIV 4000000H MOD 40H;
    Texts.WriteInt(W, 4*i, 4); Texts.Write(W, 9X); Texts.WriteString(W, mnemo[op]);
    IF op < BEQ THEN
      a := w MOD 400000H;
      IF a >= 20000H THEN DEC(a, 40000H) (*sign extension*) END ;
      Texts.Write(W, 9X); Texts.WriteInt(W, w DIV 400000H MOD 10H, 4);
      Texts.Write(W, ","); Texts.WriteInt(W, w DIV 40000H MOD 10H, 4);
      Texts.Write(W, ",")
    ELSE a := w MOD 4000000H;
      IF a >= 2000000H THEN DEC(a, 4000000H) (*sign extension*) END
    END ;
    Texts.WriteInt(W, a, 6); Texts.WriteLn(W); INC(i)
  END ;
  Texts.WriteLn(W); Texts.Append(T, W.buf)
END Decode;

```

```

BEGIN Texts.OpenWriter(W);
  NEW(boolType); boolType.form := Boolean; boolType.size := 4;
  NEW(intType); intType.form := Integer; intType.size := 4;
  mnemo[MOV] := "MOV ";
  mnemo[MVN] := "MVN ";
  mnemo[ADD] := "ADD ";
  mnemo[SUB] := "SUB ";
  mnemo[MUL] := "MUL ";
  mnemo[Div] := "DIV ";
  mnemo[Mod] := "MOD ";
  mnemo[CMPI] := "CMPI ";
  mnemo[CHKI] := "CHKI ";
  mnemo[LDW] := "LDW ";
  mnemo[LDB] := "LDB ";
  mnemo[POP] := "POP ";
  mnemo[STW] := "STW ";
  mnemo[STB] := "STB ";
  mnemo[PSH] := "PSH ";
  mnemo[BEQ] := "BEQ ";
  mnemo[BNE] := "BNE ";
  mnemo[BLT] := "BLT ";
  mnemo[BGE] := "BGE ";

```

```
mnemo[BLE] := "BLE ";
mnemo[BGT] := "BGT ";
mnemo[BR] := "BR ";
mnemo[BSR] := "BSR ";
mnemo[RET] := "RET ";
mnemo[RD] := "READ";
mnemo[WRD] := "WRD ";
mnemo[WRH] := "WRH ";
mnemo[WRL] := "WRL ";
END OSG.
```

References

- A.V. Aho, J.D. Ullman. *Principles of Compiler Design*. Reading MA: Addison-Wesley, 1985.
- F. L. DeRemer. Simple LR(k) grammars. *Comm. ACM*, 14, 7 (July 1971), 453-460.
- M. Franz. The case for universal symbol files. *Structured Programming 14* (1993), 136-147.
- S. L. Graham, S. P. Rhodes. Practical syntax error recovery. *Comm. ACM*, 18, 11, (Nov. 1975), 639-650.
- J. L. Hennessy, D. A. Patterson. *Computer Architecture. A Quantitative Approach*. Morgan Kaufmann, 1990.
- C.A.R. Hoare. Notes on data structuring.
In *Structured Programming*. O.-J. Dahl, E.W. Dijkstra, C.A.R. Hoare, Acad. Press, 1972.
- U. Kastens. *Uebersetzerbau*. Oldenbourg, 1990
- D. E. Knuth. On the translation of languages from left to right.
Information and Control, 8, 6 (Dec. 1965), 607-639.
- D.E. Knuth. Top-down syntax analysis. *Acta Informatica 1* (1971), 79-110.
- W. R. LaLonde, et al. An LALR(k) parser generator.
Proc. IFIP Congress 71, North-Holland, 153-157.
- J.G.Mitchell, W. Maybury, R. Sweet. Mesa Language Manual.
Xerox Palo Alto Research Center, Technical Report CSL-78-3.
- P. Naur (Ed). Report on the algorithmic language Algol 60.
Comm. ACM, 3 (1960), 299-314, and *Comm. ACM*, 6, 1 (1963), 1-17.
- P. Rechenberg, H. Mössenböck. *Ein Compiler-Generator für Mikrocomputer*. C. Hanser, 1985.
- M. Reiser, N. Wirth. *Programming in Oberon*. Wokingham: Addison-Wesley, 1992.
- N. Wirth. The programming language Pascal. *Acta Informatica 1* (1971)
- N. Wirth. Modula - A programming language for modular multiprogramming.
Software - Practice and Experience, 7, (1977), 3-35.
- N. Wirth. What can we do about the unnecessary diversity of notation
for syntactic definitions? *Comm. ACM*, 20, (1977), 11, 822-823.
- N. Wirth. *Programming in Modula-2*. Heidelberg: Springer-Verlag, 1982.
- N. Wirth and J. Gutknecht. *Project Oberon*. Wokingham: Addison-Wesley, 1992.